

Speedy Computer

# GIOCHI di spionaggio brivido e mistero



GRUPPO EDITORIALE JACKSON











# GIOCHI di spionaggio brivido e mistero



## Sommario

3 Giochi  
di Spionaggio

21 Giochi  
del Brivido

39 Giochi  
del Mistero





# GIOCHI DI SPIONAGGIO

Jenny Tyler e Chris Oxlade



## Indice

- 4 I programmi dei giochi di spionaggio
- 5 Occhi da Spia
- 6 Faro nella notte
- 8 Robot Spia
- 10 Prova di abilità
- 12 Cifra Messaggi Segreti
- 14 Appuntamento di Spie
- 18 Codice Morse
- 19 Soluzioni dei quesiti

## I programmi dei giochi di spionaggio

I programmi sono scritti in una versione standard del BASIC, e ci sono le linee di conversione per la maggior parte delle principali marche di home-computer. Cerca sulla sinistra del programma il simbolo del tuo computer e poi vai a cercare la versione adatta di quella linea nella lista delle modifiche. Questi sono i simboli per le varie marche di home-computer:

- ▲ **Commodore 64 e VIC 20**
- ★ **BBC ed Electron**
- **Spectrum**
- **Apple**
- **TRS-80**  
(versione di BASIC esteso)

## Qualcosa sui giochi

I giochi sono molto semplici. Il loro scopo è quello di darti una mano affinché tu ti abitui al tuo computer e al linguaggio BASIC, introducendo i listati, correggendoli, e guardando come funzionano. I programmi non contengono né grafica né suono, dato che questi variano molto da computer a computer, ma se vuoi, puoi provare ad aggiungerli tu.

Puoi fare tanti cambiamenti e adattamenti ai giochi quanti ne vuoi: ci sono suggerimenti sui modi di farlo vicino ad ogni programma, e puoi sperimentare anche con idee tue. In questo modo potrai usare i giochi contenuti in questo libro come una base di partenza per creare i tuoi programmi che saranno più lunghi e complicati.

*"Robot-Spia" è stato scritto da Adrian Hall; progettato da Roger Priddy; illustrato da Sue Walliker, Rob McCaig e Martin Newton*

## Come introdurre i programmi e come farli "girare"

Ricordati che anche programmi corti possono essere molto difficili da copiare correttamente e possono richiedere molto tempo. Controlla ogni linea mentre la scrivi, dato che è molto facile commettere errori anche se si è molto esperti. Dopo aver introdotto tutto il listato, ricontrollalo per accertarti di non avere saltato linee, spazi o segni di punteggiatura.


Per iniziare il gioco, dai il RUN. Leggi prima la presentazione del gioco, in modo da avere un'idea di ciò che devi fare prima di iniziare. Se il programma non funziona correttamente, è abbastanza probabile che ci sia un errore da qualche parte, perciò dai il LIST e ricontrollalo nuovamente.

Quando la partita è finita, può darsi che il computer ti chieda se vuoi farne un'altra, oppure ti comunichi qualcosa come BREAK IN 200, nel qual caso dovrai dare il RUN per giocare ancora.

## Come cambiare la velocità

Alcuni giochi dipendono sia dalla tua prontezza di riflessi che dalla velocità del tuo computer. Può darsi che tu voglia regolare la velocità, in questo caso troverai le istruzioni accanto al listato del programma.





# Occhi da Spia

Se pensi di essere una buona spia, prova questo gioco. Il computer visualizzerà i numeri da 1 a 9 sul tuo schermo. Osservali con occhi di falco mentre premi un tasto (uno qualsiasi va bene): uno dei numeri si è spostato, ma quale? Quando pensi di sapere qual'è, premi un tasto di nuovo e dillo al computer.

## Come funziona

10: Riserva spazio per l'immagazzinamento delle posizioni dei numeri.

20: Azzerà il punteggio.

30-60: Sceglie le posizioni dei numeri.

70: Va alla subroutine per la visualizzazione dei numeri.

80: Va alla subroutine che attende la pressione di un tasto.

90-100: Va alla subroutine che sceglie un numero, e lo chiama M.

110: Decide lo spostamento verso sinistra o verso destra.

120: Va alla subroutine per la visualizzazione dei numeri.

130: Va alla subroutine che attende la pressione di un tasto.

140-150: Pulisce lo schermo e visualizza il messaggio.

160: Ti chiede un numero.

170: Se il numero che hai introdotto non è esatto, il programma va al messaggio della linea 260.

180-190: Pulisce lo schermo e stampa il messaggio.

200-210: Aggiorna il punteggio e lo visualizza sullo schermo.

220-230: Stampa il messaggio e va alla subroutine che attende la pressione di un tasto.

240: Torna indietro per il giro successivo.

250-300: Stampa messaggi se hai sbagliato e ti chiede se vuoi riprovare.

310-330: Subroutine che attende la pressione di un tasto.

340-350: Subroutine per la scelta casuale di un numero tra 1 e 9.

360-400: Subroutine che stampa i numeri sullo schermo nella posizione prescelta.


```
3 REM OCCHI DA SPIA
5 REM VERSIONE GENERALE
10 DIM X(9):DIM Y(9)
20 LET P=0
30 FOR I=1 TO 9
40 GOSUB 340:LET X(I)=N+3
50 GOSUB 340:LET Y(I)=N+3
60 NEXT I
70 GOSUB 360
80 GOSUB 310
90 GOSUB 340
100 LET M=N:GOSUB 340
110 LET X(M)=X(M)+SGN (N-5.1)
120 GOSUB 360
130 GOSUB 310
140 CLS:PRINT
150 PRINT "QUAL'E' IL NUMERO CHE SI E' MOSSO"
160 INPUT A
170 IF A<0 THEN GOTO 250
180 CLS:PRINT
190 PRINT "BEN SPIATO!"
200 LET P=P+1
210 PRINT "ORA HAI ";P;" PUNTI"
220 PRINT:PRINT "PREMI UN TASTO"
230 GOSUB 310
240 GOTO 30
250 CLS:PRINT "SBAGLIATO!"
260 PRINT "LA RISPOSTA ESATTA ERA ";M
270 PRINT "HAI OTTENUTO NO.";P;" PUNTI"
275 PRINT "FINE DELLA PARTITA.":PRINT:PRINT
280 PRINT "UN'ALTRA PARTITA? (S/N)"
290 INPUT A$:IF A$="S" THEN RUN
300 STOP
310 LET I$=INKEY$
320 IF I$="" THEN GOTO 310
330 RETURN
340 LET N=INT (RND (1)*9)+1
350 RETURN
360 CLS
370 FOR I=1 TO 9
380 PRINT TAB(X(I),Y(I));STR$(I)
390 NEXT I
400 RETURN
```



Prova a inventare una routine di punteggio massimo per questo gioco.

## Linee di conversione

- 140, 180, 250, 360 Sostituisci CLS con HOME
- ▲ 140, 180, 250, 360 Sostituisci CLS con PRINT CHR\$(147)
- ★ 310 LET I\$=INKEY\$(0)
- 310 I\$="" IF PEEK(-16384)>127 THEN GET I\$
- ▲ 310 GET I\$
- 340 Sostituisci RND(1) con RND
- 340 LET N=RND(9)
- 380 PRINT AT Y(I),X(I);STR\$(I)
- 380 VTAB(Y(I)):HTAB(X(I)):PRINT STR\$(I)
- ▲ 380 PRINT CHR\$(19):FOR LL=1 TO Y(I):PRINT:PRINT TAB(X(I));STR\$(I)
- 380 PRINT @ Y(I)\*32+X(I),STR\$(I);



Stanotte stessa toccherà a te intraprendere una missione estremamente segreta e pericolosissima. Devi attraversare un'area posta sotto accurata sorveglianza in territorio nemico, evitando di essere illuminato dal gigantesco e potente riflettore. Ci sono massi, cespugli e altri ostacoli dietro cui nascondersi, ma non ti sarà data una seconda possibilità se ti fai scoprire. Una volta completata con successo una missione, ce n'è un'altra da intraprendere, che sarà più difficile. Persisti, contiamo tutti sul tuo successo. Usa il tasto M per spostarti verso destra e N per andare verso sinistra. Per completare la missione, devi attraversare il campo da sinistra a destra e poi tornare indietro.

```

3 REM FARO NELLA NOTTE
4 REM VERSIONE GENERALE
10 GOSUB 450
20 LET A=1:LET G=0:LET S=0
30 CLS
40 LET X=0:LET Y=12:LET B=A*(A)
50 GOSUB 380
60 LET F=0:LET N=0:LET NN=0:GOSUB 340
70 LET L=0:LET C=0:LET TC=10:LET C1=0
80 LET I$=INKEY$
90 IF I$="N" THEN LET NN=NN-1
100 IF I$="M" THEN LET NN=NN+1
110 IF NN>19 THEN LET NN=19
120 IF NN<0 THEN LET NN=0
130 IF NN=19 AND F=0 THEN LET F=1
140 IF NN=0 AND F=1 THEN LET F=2
150 GOSUB 340
160 IF NN=NN THEN LET S=S+1
170 LET NN=NN:LET G=G+1
180 GOSUB 400
190 IF MID$(A$(A),N+1,1)=" " AND L=1 THEN GOTO 240
200 FOR T=1 TO 50:NEXT T
210 IF F<2 THEN GOTO 80
220 LET A=A+1:IF A=8 THEN LET A=7
230 GOTO 30
240 LET X=4:LET Y=1:LET B$="SEI STATO VISTO"
250 GOSUB 380:PRINT
260 PRINT " IL TUO PUNTEGGIO: ";INT ((A-1+S/G)*100)
270 PRINT:PRINT "UN'ALTRA PARTITA ? (S/N)"
280 INPUT C$:IF C$="S" THEN RUN
290 PRINT "CIAO..." :STOP
300 LET Y=3:LET X=10:LET B$="*"
    
```

## Come funziona

10: Va alla subroutine per leggere i dati.

20: Prepara le variabili.

40-50: Visualizza gli ostacoli.

60-70: N è la posizione del giocatore; NN è la nuova posizione.

80: Controlla se è stato premuto un tasto.

90: Se il tasto premuto è N (sinistra), la nuova posizione sarà N meno 1; se è M (destra), sarà N più 1.

100-120: Impedisce al giocatore di oltrepassare i limiti dell'area.

130-140: F vale 1 quando si raggiunge il lato destro dello schermo e 2 ancora il lato sinistro.

150: Va alla subroutine che muove il giocatore.

160: Se il giocatore ha mosso, S viene aumentato.

170: Pone la posizione del giocatore uguale a quella nuova e aumenta G di 1. Il valore di G cresce col passare del tempo, indipendentemente dal fatto che il giocatore muova o meno.

180: Va alla subroutine che accende e spegne il faro.

190: Controlla se il giocatore si trova sotto uno spazio mentre il faro è acceso, e in questo caso va al messaggio di sconfitta.

200: Pausa.

210: Se il giocatore non è tornato al lato sinistro, torna a vedere se è stato premuto un altro tasto.

Sei capace di fare in modo che il faro resti acceso più a lungo.

Prova a sostituire i simboli = che raffigurano gli ostacoli con simboli grafici che rappresentano alberi, massi, edifici, o qualsiasi altra cosa ti venga in mente.





```

310 GOSUB 380:RETURN
320 LET X=10:LET Y=3:LET B$=" "
330 GOSUB 380:RETURN
340 LET X=N:LET Y=13:LET B$=" "
350 GOSUB 380
360 LET X=NN:LET B$="S"
370 GOSUB 380:RETURN
□●●▲380 PRINT TAB(X,Y);B$
390 RETURN

```



Aggiungi un suono d'allarme per quando il giocatore è scoperto.

```

400 IF L=1 THEN LET C=C+1
□■410 IF C=TC THEN LET L=0:LET C=0:LET TC=INT (RND (1)*8+(12-A)):GOSUB 320
420 IF L=0 THEN LET C1=C1+1
□■430 IF C1=TC THEN LET L=1:LET C1=0:LET TC=INT (RND (1)*10+(8-A)):GOSUB 300
440 RETURN
□450 DIM A$(7)
460 FOR I=1 TO 7:READ A$(I):NEXT I
470 RETURN
480 DATA "== == == == == == =="
490 DATA "== == == == == == =="
500 DATA "== == == == == == =="
510 DATA "== == == == == == =="
520 DATA "== == == == == == =="
530 DATA "== == == == == == =="
540 DATA "== == == == == == =="

```

Puoi spostare gli ostacoli modificando le linee di DATA.

220: Aumenta il livello di difficoltà.

230: Torna a visualizzare gli ostacoli del nuovo livello.

240-290: Stampa il messaggio di sconfitta e il punteggio; chiede al giocatore se vuole fare un'altra partita.

300-310: Visualizza il faro.

320-330: Spegne il faro.

340-370: Subroutine che modifica la posizione del giocatore.

380-390: Subroutine per la visualizzazione su schermo in posizione X, Y.

400-440: Subroutine che determina i tempi del riflettore.

450-470: Subroutine di lettura dati.

480-540: Linee contenenti i dati.

Potresti rendere più facile il gioco, dando una seconda possibilità al giocatore. Come si potrebbe fare?



## Linee di conversione

- 30 HOME
- ▲ 30 PRINT CHR\$(147)
- ★ 80 LET IS=INKEY\$(0)
- ▲ 80 GET IS
- 80 IS="":IF PEEK (—16384)>127 THEN GET IS
- 190 IF A\$(A,N+1)=" " AND L=1 THEN GOTO 240
- 380 PRINT AT Y,X;B\$
- 380 VTAB(Y):HTAB(X+1):PRINT B\$
- ▲ 380 PRINT CHR\$(19):FOR LL=1 TO Y:PRINT:NEXT:PRINT TAB(X);B\$
- 380 PRINT @ Y\*32+X,B\$;
- 410, 430 Sostituisci RND(1) con RND
- 410, 430 Sostituisci RND(1) con RND(0)
- 450 DIM A\$(7, 20)
- ★ La velocità è controllata dalla linea 200
- Sul BBC e sull'Electron, modificala in FOR T=1 TO 150: NEXT T



Sei tu ad avere il controllo di Robot Spia: un apparecchio di pedinamento unico al mondo, controllato a distanza, che segue di nascosto gli agenti nemici. Ti vengono comunicati i dati relativi agli spostamenti di un agente per le strade, sia che egli giri a destra o a sinistra, e tu devi replicare esattamente questi movimenti quando mandi i segnali al robot-spia, per poterlo mantenere in contatto con l'agente.

Sfortunatamente l'agente sa che Robot Spia lo sta seguendo, e ti rende il compito sempre più difficile man mano che il gioco va avanti. Sei capace di stargli alle costole, oppure ti farà perdere le sue tracce?

Fai in modo che il computer chieda il nome del giocatore quando è superato il punteggio massimo?



```
3 REM ROBOT SPIA
5 REM VERSIONE GENERALE
10 LET SP=0.3
●▲20 CLS
30 LET H=0
40 LET K=0:LET S=0:LET I=1
50 PRINT ">>> ROBOPIA <<< PRINT
```

```
60 PRINT "PREMI I PER INIZIARE"
●●▲70 LET C$=INKEY$
80 IF C$="I" THEN GOTO 70
●▲90 CLS:PRINT:PRINT
100 IF K=5 THEN LET L=L+1:LET K=0
110 LET M$=""
120 LET C$=""
130 FOR T=1 TO L
□140 LET N=RND(1)
150 IF N<0.5 THEN PRINT "SINISTRA" LET M=M$+"S"
160 IF N<0.5 THEN PRINT "DESTRA" LET M=M$+"D"
170 NEXT T
180 LET K=K+1
190 FOR D=1 TO 200*L*SP
●●▲200 LET C$=INKEY$
210 NEXT D
●▲220 CLS:PRINT:PRINT
230 LET E=0
240 PRINT "QUALI ERANO LE DIREZIONI"
250 FOR T=1 TO L
260 PRINT "?:
●●▲270 LET C$=INKEY$
280 IF C$="S" AND C$="D" THEN GOTO 270
290 PRINT C$
□300 IF C$=MID$(M$,T,1) THEN LET E=1
310 NEXT T
320 IF E=0 THEN LET S=S+L
330 IF S>H THEN LET H=S
340 IF E=1 THEN PRINT "L'HAI PERSO! CHE RAZZA DI SPIA!"
350 PRINT "PUNTEGGIO = ":S
360 PRINT "PUNTEGGIO MASSIMO = ":H
370 IF E=1 THEN GOTO 40
380 FOR D=1 TO 500*SP
390 NEXT D
400 GOTO 90
```

## Linee di conversione

- 20, 90, 220 Sostituisci CLS con HOME
- ▲ 20, 90, 220 Sostituisci CLS con PRINT CHR\$(147)
- 70, 200, 270 LET C\$=INKEY\$(0)
- 70, 200, 270 C\$="":IF PEEK(-16384)>127 THEN GET C\$
- ▲ 70, 200, 270 GET C\$
- 140 LET N=RND
- 140 LET N=RND(0)
- 300 IF C\$ < > M\$(T) THEN LET E=1

## Come funziona

30-40: Prepara le variabili. L è il numero di parole sullo schermo; K tiene conto di quante volte apparse sullo schermo L parole.

60-80: Visualizza il messaggio e attende la pressione di un tasto per iniziare il gioco.

90: Pulisce lo schermo e lascia 2 linee vuote.

100: Se K è 5, il valore di L viene aumentato di 1 e K viene azzerato.

110-120: Prepara due variabili-stringa vuote.

130-170: Per L volte ripete la sequenza per la scelta casuale di direzione (destra o sinistra), la visualizza e l'immagazzina in M\$.

180: Aumenta K di 1.

190-210: Pausa per far vedere le parole al giocatore. Più sono le parole e più lunga sarà la pausa.

230: Variabile "flag" di errore: viene attivata (=1) se il giocatore sbaglia.

250: Inizio del loop di attesa e controllo della risposta.

260-270: Ti indica che è pronto, aspetta che venga premuto un tasto e lo mette in C\$.

280: Se il tasto premuto non è né D né S, torna indietro ad aspettarne un altro.

300: Controlla se la lettera in C\$ è uguale a quella corrispondente di M\$ e attiva la variabile "flag" di errore se non lo è.

320: Aumenta il punteggio se non vi sono errori.

330: Aggiorna il valore del punteggio massimo, se necessario.

340: Visualizza il messaggio d'errore, se ce n'è stato uno.

350-360: Stampa i punteggi.

370: Torna indietro per un'altra partita.

380-390: Fa una pausa per mostrare il punteggio.

400: Torna indietro per il giro successivo.

La linea 10 controlla la velocità del gioco. Modifica questo numero come segue: Spectrum 0.2, VIC 20 0.4, Electron 0.6, BBC 1, Apple 0.1.

Qualunque computer tu stia usando, minore è il numero e più veloce sarà il gioco. Prova ad aumentare la velocità e vedi come ti riesce il gioco.



# Prova di abilità

Come nuova recluta alla Scuola di Spie, devi superare molte prove di abilità per far carriera. Inizi dal livello più basso, quello di Aspirante Spia (grado 5°). Il tuo scopo è quello di raggiungere il massimo e diventare Capospia e forse anche di ottenere il massimo dei riconoscimenti: il trofeo di Superspia. In ogni prova, ti verranno visualizzate sullo schermo dieci posizioni; poi ti verranno dati dieci numeri tra l'1 e il 100. Il tuo compito è quello di porre questi dieci numeri in ordine crescente, col numero più basso alla posizione 1 e il più alto alla 10. Puoi scartare alcuni numeri, se non si possono inserire, premendo S. Il numero di volte che puoi usare S è pari al numero del tuo livello.

```
3 REM PROVA DI ABILITA' PER SPIE
5 REM VERSIONE GENERALE
10 DIM N(10):DIM N$(5)
20 CLS
30 GOSUB 510
40 LET W$=""
50 LET D=5
60 LET G=0
70 FOR I=1 TO 10:LET N(I)=0:NEXT I
80 LET I=1
90 GOSUB 430
100 LET M=INT (RND (1)*99+1)
110 PRINT:PRINT "DOVE METTERESTI ":M
120 PRINT:INPUT P$
130 IF P$="S" AND G<D THEN LET G=G+1:GOTO 90
140 IF P$="S" THEN PRINT "NON PUOI!":GOTO 120
150 LET P=VAL (P$)
160 IF P<1 OR P>10 THEN GOTO 120
```

## Come funziona

10: Riserva spazio in memoria per gli array.

20: Pulisce lo schermo.

30: Va alla subroutine che legge i dati e li pone nell'array N\$(5) tramite READ.

40: W\$ è una stringa vuota.

50: Pone il numero di grado uguale a 5 per iniziare.

60: G corrisponde al numero di volte che hai scartato un numero.

70: Mette zero in ognuna delle dieci posizioni per la partenza.

80: Conta i numeri già messi in posizione dal giocatore.

90: Va alla subroutine che visualizza il tuo grado, i numeri da 1 a 10 e i numeri già posizionati.

100: Sceglie un numero tra 1 e 100.

110-120: Stampa il numero e ti chiede dove lo vuoi mettere.

130-140: Controlla se hai premuto S e poi controlla se puoi scartare ancora.

150: Calcola il valore numerico del tuo input.

160: Lo rifiuta se non è compreso tra 1 e 10.

Sei capace di mettere  
15 posizioni invece di  
10?

Cambia i gradi  
se vuoi.

Hai la stoffa  
della  
Superspia?



```

170 IF N(P)>0 THEN PRINT "E' GIA' PIENO":GOTO 120
180 LET N(P)=M
190 LET F=0
200 FOR L=P TO 10
210 IF N(L)<M AND N(L)<0 THEN LET F=1
220 NEXT L
230 FOR L=1 TO P
240 IF N(L)>M AND N(L)<0 THEN LET F=1
250 NEXT L
260 IF F=1 THEN GOTO 360
270 LET I=I+1:IF I<11 THEN GOTO 90
280 LET D=D-1:IF D=0 THEN GOTO 330
290 PRINT "PROMOSSO AL GRADO ";D
300 PRINT:PRINT "ORA SEI UNA ";N$(D)
310 LET W$=""
320 GOTO 400
330 PRINT "BRAVISSIMO - HAI RAGGIUNTO"
340 PRINT "IL GRADO DI SUPERSPIA"
350 STOP
360 PRINT "SBAGLIATO! NON SEI ABBASTANZA BRAVO"
370 PRINT:PRINT "SEI ANCORA UNA ";
380 PRINT N$(D)
390 LET W$="ANCORA"
400 PRINT:PRINT "VUOI PROVARE ANCORA? (S/N)"
410 INPUT A$:IF A$="S" THEN GOTO 60
420 STOP
●▲430 CLS
440 PRINT:PRINT "SEI ";W$;" UNA ";N$(D)
450 PRINT
460 FOR J=1 TO 10
470 PRINT J:
480 IF N(J)>0 THEN PRINT N(J):
490 PRINT:NEXT J
500 RETURN
510 FOR I=1 TO 5:READ N$(I)
520 NEXT I
530 RETURN
540 DATA "CAPOSPIA", "SPIA", "VICESPIA"
550 DATA "AIUTO-SPIA","ASPIRANTE SPIA"

```

**170:** Lo rifiuta se la posizione è già occupata.

**180:** Mette il numero nella posizione che hai scelto.

**190-220:** Controlla tutte le posizioni più alte di quella che hai scelto per vedere se c'è già un numero minore in una di esse; in questo caso, attiva F.

**230-250:** Controlla tutte le posizioni inferiori a quella che hai scelto per vedere che non ci sia già un numero più alto in una di esse; se c'è, attiva F.

**260:** Se la posizione è sbagliata, va avanti nel programma per dirtelo.

**270:** Aumenta il valore del contatore e torna indietro per un altro numero.

**280-300:** Modifica S per il grado successivo e stampa il messaggio. (Se S vale 0, va alla Superspia).

**310:** W\$ è una stringa vuota.

**320:** Va al messaggio "riprova".

**330-350:** Messaggio Superspia.

**360-380:** Stampa il messaggio di sconfitta.

**390:** Pone W\$ uguale ad "ANCORA".

**400-420:** Aspetta di vedere se vuoi giocare ancora; se non vuoi, ferma il gioco.

**430-500:** Subroutine che visualizza il messaggio e i numeri.

**520-530:** Subroutine per la lettura dei nomi dei gradi che verranno messi in N\$.

**540-550:** Dati.

## Linee di conversione

- 10 DIM N(10):DIM N\$(5,16)
- 20,430 Sostituisci CLS con HOME
- ▲ 20,430 Sostituisci CLS con PRINT CHR\$(147)
- 100 LET M = INT (RND\*99 + 1)
- 100 LET M = RND(99)



# Cifra Messaggi Segreti

Usa questo programma per mandare dei messaggi cifrati ai tuoi amici. Essi avranno bisogno di un computer per decifrarli, ma non necessariamente dello stesso tipo di quello che hai tu. (Avranno però bisogno di una copia del programma). Essi potranno decifrare i tuoi messaggi e mandarti i loro.



Potresti organizzare una Rete di Spie con i tuoi amici che hanno un computer.

```

3 REM MESSAGGI CIFRATI
5 REM VERSIONE GENERALE
●▲10 CLS:PRINT:PRINT "CIFRA-MESSAGGI SEGRETI"
20 PRINT "=====
30 PRINT:PRINT "COSA VUOI FARE: "
40 PRINT
50 PRINT "          1. CIFRARE UN MESSAGGIO"
60 PRINT "OPPURE  2. DECIFRARE UN MESSAGGIO"
70 PRINT:PRINT
80 PRINT "INSERISCI UN NUMERO ":INPUT A
90 IF A=1 THEN GOSUB 120
100 IF A=2 THEN GOSUB 210
110 GOTO 30
120 LET C$="CIFRATO":GOSUB 400
130 LET X=INT (RND (!)*25+1)
140 LET M$="F"+M$
150 GOSUB 310:GOSUB 420
160 LET M$=CHR$ (X+64)+M$
170 IF LEN (M$)/2=INT (LEN (M$)/2) THEN GOSUB 450
180 PRINT "IL MESSAGGIO CIFRATO E':"
190 PRINT M$
200 RETURN
210 LET C$="DECIFRATO":GOSUB 400
220 IF LEN (M$)/2=INT (LEN (M$)/2) THEN GOSUB 450
230 LET K$=LEFT$ (M$,1)
240 LET M$=RIGHT$ (M$,LEN (M$)-1)
250 LET X=ASC (K$)-64
260 LET X=X:GOSUB 420
    
```

## Come funziona

10-80: Visualizza il "menu" e ti chiede se vuoi cifrare o decifrare.

90-100: Va alle relative subroutine.

110: Torna al menu.

120-200: SUBROUTINE DI CODIFICA.

120: Va alla subroutine per la richiesta di messaggio.

130: Sceglie l'ammontare dello spostamento della lettera all'interno del codice e lo immagazzina in X.

140: Aggiunge una lettera nel caso in cui la prima parola sia formata da un solo carattere.

150: Va alle subroutine che spostano le singole lettere del messaggio di X lettere più in là nell'alfabeto e ne rovescia l'ordine.

160: Aggiunge una lettera al messaggio per dire alla subroutine di decodifica qual'è il valore di X.

170: Controlla se il messaggio ha un numero di caratteri pari. In questo caso va alla subroutine per scambiare fra loro le lettere di ogni coppia di caratteri.

## 210-300: SUBROUTINE DI DECODIFICA

210: Va alla subroutine per chiedere il messaggio da decifrare.

220: Se il messaggio contiene un numero di caratteri pari, scambia fra loro le lettere di ogni coppia.

230-240: Toglie la prima lettera. (E quella aggiunta alla linea 160 per mostrare l'ammontare dello spostamento).

250: Calcola il valore di X.

260: Cambia la direzione dello spostamento e va alla subroutine che rovescia l'ordine delle lettere componenti il messaggio.

TEWSYTFYN VZN TWWW  
JQN LHFJWKH

NBRJO SFGG CHO  
GHWC IBZQIOTE

WXRFDXXI MFEZ  
LFEZZBS IBRNC

KBPQC VTC NRVCC  
WPC CGHE

UNDMJW V ZMZY  
ZQZY DN IJIA

NOB GDP TWXUR



```

□270 GOSUB 310:M$=RIGHT$(M$,LEN(M$)-1)
280 PRINT "IL MESSAGGIO DECIFRATO E':"
290 PRINT M$
300 RETURN
310 LET N$=""
320 FOR I=1 TO LEN(M$)
□330 LET Q$=MID$(M$,I,1):LET N=ASC(Q$)
340 IF N=32 THEN GOTO 380
350 LET N=N+X
360 IF N>90 THEN LET N=N-26
370 IF N<65 THEN LET N=N+26
380 LET N$=N$+CHR$(N):NEXT I
390 LET M$=N$:RETURN
400 PRINT "QUAL'E' IL MESSAGGIO CHE DEVE ESSERE ";C$
410 INPUT M$:RETURN
420 LET N$="":FOR I=LEN(M$) TO 1 STEP -1
□430 LET N$=N$+MID$(M$,I,1):NEXT I
440 LET M$=N$:RETURN
450 LET N$="":LET L=LEN(M$)
460 FOR I=1 TO LEN(M$)-1 STEP 2
□470 LET N$=N$+MID$(M$,I+1,1)
□480 LET N$=N$+MID$(M$,I,1)
490 NEXT I:LET M$=N$:RETURN

```

Una buona spia controlla sempre che i suoi messaggi siano correttamente decifrabili, prima di spedirli.



270: Va alla subroutine che sposta le lettere nell'alfabeto. Toglie la prima lettera.

### 310-390: SUBROUTINE DI SPOSTAMENTO.

310: Pone il messaggio in N\$.

320: Inizio di un loop che toglie le lettere da N\$ una ad una.

330: Trasforma le lettere nel numero di codice del computer.

340: Se si tratta di uno spazio, lo lascia invariato.

350: Aggiunge il valore di X al codice della lettera.

360-370: Modifica il codice della lettera se esso è superiore al valore di Z o inferiore al valore di A.

380: Aggiunge una nuova lettera a N\$ e ritorno all'inizio del loop.

390: Quando tutte le lettere sono state riportate al loro valore originario, il nuovo messaggio viene posto in M\$.

400-410: Subroutine di richiesta messaggio.

420-440: Subroutine che rovescia l'ordine delle lettere.

450-490: Subroutine che prende le lettere dal messaggio a due a due e scambia di posto fra loro le lettere di ogni coppia.

## Linee di conversione

- 10 Sostituisci CLS con HOME
- ▲ 10 Sostituisci CLS con PRINT CHR\$(147)
- 130 LET X=INT(RND\*25)+1
- 130 LET X=RND(25)
- 230 LET K\$=M\$(1)
- 240 LET M\$=M\$(2 TO)

- 250 LET X=CODE(K\$)-64
- 270 GOSUB 310:LET M\$=M\$(2 TO)
- 330 LET Q\$=M\$(I):LET N=CODE(Q\$)
- 430 LET N\$=N\$+M\$(I):NEXT I
- 470 LET N\$=N\$+M\$(I+1)
- 480 LET N\$=N\$+M\$(I)



DUIJV YZV ODGZ  
ZMIYMZ K VIDDZAQ

CXRRLMRY QAGKMG  
YRPMLLADG

KBTUFGGPIVK FTKNV  
CNKCIK KEHF

EINX FM NQ NSFNU  
NK

Sei capace di decifrare che cosa dicono?



# Appuntamento di Spie

Hai una difficile missione da svolgere, per cui leggi attentamente queste istruzioni. Devi ritirare una valigia da una cassetta alla stazione, consegnarla al tuo contatto e poi ritornare all'aeroporto prima che parta l'ultimo aereo. Il tuo computer di dirà l'ora, dove si troverà il tuo contatto e a che ora sarà lì. Devi lasciare un messaggio in quel posto, prima che egli arrivi, dicendogli dove e quando vi incontrerete per la consegna del pacco. Devi scoprire la parola d'ordine prima di incontrarlo e accertarti di non aver un ritardo superiore ai 15 minuti. Prima di poter ritirare la valigia, devi trovare la chiave e anche il numero della cassetta. Sfortunatamente, la chiave è nelle mani di agenti nemici, che hanno il quartier generale all'Albergo. Devi trovare una spia nemica e seguirla, sperando che lasci cadere la chiave (e che non ti veda). La mappa illustra i posti dove puoi andare, e la lista sotto riporta le parole che potrai usare nel gioco.



## Parole che puoi usare

**ORE** - Ti dice che ora è.


**ANDARE** - Ti chiederà dove. Puoi andare in qualsiasi posto segnato nella mappa.

**DIRE** - La parola d'ordine.

**ESAMINARE** - Qualsiasi cosa. (Esamina la chiave per vederne il numero).

**LEGGERE** - Un messaggio.

**APRIRE** - La cassetta.



**SEGUIRE** - La spia nemica.

**ASPETTARE** - Finché vuoi.

**LASCIARE** - Un messaggio.

**CERCARE** - Ovunque (per trovare la chiave).


**AIUTO** - Ti ricorda l'ora e il posto dell'appuntamento.

Inoltre puoi usare i nomi scritti in stampatello sulla mappa.

Se vuoi, cambia le parole d'ordine.

Puoi cambiare anche i nomi dei posti, se vuoi.





Se usi il VIC 20 ti occorre per questo gioco memoria extra.

```

3 REM APPUNTAMENTO DI SPIE
5 REM VERSIONE GENERALE
■ □ 10 DEF FN A(X)=INT (RND (1)*X)+1
20 GOSUB 1010:GOSUB 1150
30 LET P=1:LET Y=0:LET X=1:LET EP=0
40 LET U=0:LET NM=0
50 LET S$=""
60 LET B$=""
● ▲ 70 CLS:PRINT:PRINT "APPUNTAMENTO DI SPIE"
80 PRINT "===== "
90 LET NM=NM+1
100 PRINT:PRINT
110 PRINT B$:PRINT
120 PRINT:PRINT "TI TROVI PRESSO ":
130 AR=P:GOSUB 1360:PRINT R$(P)
140 IF EP=P THEN PRINT "L'AGENTE NEMICO E' QUI":LET NE=NE+1
150 IF EP<P THEN LET NE=0
160 IF P=MP AND F(1)=0 THEN PRINT "C'E' QUI UN MESSAGGIO PER TE"
170 LET F(3)=0
180 LET TI=H+M/100

```

## Come funziona

**10:** Prepara una funzione per la scelta casuale di un numero tra 1 e X.

**20:** Va alla fine del programma per leggere i dati e preparare il gioco (inizializzazione).

**30:** P, Y e X registrano la posizione del giocatore. EP è la posizione del nemico.

**60:** B\$ è per i messaggi del computer, rivolti al giocatore.

**90:** Aumenta NM (numero di mosse).

**140:** NE tiene conto di quanto tempo il giocatore resta nello stesso posto dove si trova l'agente nemico.

**180:** Calcola l'ora (H=ore, M=minuti).

190-200 Controlla se la missione è stata portata a termine.

210-270: Prende la parola introdotta dall'utente e la confronta con le parole in memoria. V è il numero della parola corrispondente.

```
190 IF F(4)=1 AND R$(P)=S$ AND UC=T1 AND U+.15>T1 THEN PRINT T$:LET F(3)=1
200 IF P=1 AND KCFH AND F(7)=1 THEN GOTO 890
210 PRINT:PRINT:PRINT "COSA VUOI FARE"
220 LET B$=""
230 INPUT I$
240 LET V=0:FOR I=1 TO 11
250 IF I$=V$(I) THEN LET V=I
260 NEXT I
270 IF V=0 THEN LET V=12
```

280-290: Controlla il valore di NE per vedere se il giocatore è stato visto dal nemico.

300: Va alla subroutine relativa alla parola introdotta.

310: Aggiorna l'ora.

330: Controlla se il tempo è scaduto.

340: Fa ritornare il nemico all'albergo.

360-410: SUBROUTINE "ANDARE"

```
280 IF NE=3 AND FN A(10)>3 AND VC>1 THEN LET B$="L'AGENTE NEMICO TI HA VISTO!":G
OTO 70
290 IF NE=4 THEN PRINT "SEI STATO CATTURATO!":STOP
300 ON VGOSUB 360,420,490,540,570,640,710,730,780,810,820,870
310 LET M=M+DT:IF M>59 THEN LET M=M-60:LET H=H+1
320 IF F(2)=1 AND HD=CH THEN LET F(4)=1
330 IF H=FH THEN GOTO 880
340 IF FN A(10)>9 THEN LET EP=10
350 GOTO 70
360 PRINT:PRINT "DOVE":INPUT N$
370 LET NP=0:FOR I=1 TO 20
380 IF N$=R$(I) THEN LET NP=I
390 NEXT I:IF NP=0 THEN GOTO 360
400 GOSUB 950
410 LET P=NP:RETURN
420 LET DT=5
430 PRINT:PRINT "COSA DEVO DIRE":INPUT Q$
```

420-490 SUBROUTINE "DIRE"

500-540: SUBROUTINE "ESAMINARE"

550-570: SUBROUTINE "LEGGERE"

```
440 IF EP=P THEN LET B$="HAI ATTIRATO L'AGENTE NEMICO!":RETURN
450 IF F(3)=0 THEN LET B$="NESSUNO TI SENTE":RETURN
460 IF Q$<P$ THEN LET B$="IL CONTATTO TI HA IGNORATO!":RETURN
470 IF F(6)=1 THEN LET B$="IL CONTATTO E' STATO EFFETTUATO - HA PRESO LA BORSA!"
480 LET F(7)=1:RETURN
490 LET DT=5
500 PRINT:PRINT "COSA VUOI ESAMINARE":INPUT Q$
510 IF Q$="BORSA" THEN LET B$="TOP SECRET!":RETURN
520 IF Q$="CHIAVE" THEN LET B$="PORTA IL NUMERO "+STR$ (NL):RETURN
530 LET B$="NIENTE DI SPECIALE!":RETURN
540 IF P<MP OR F(1)=1 THEN LET B$="NON C'E' NIENTE DA LEGGERE!":RETURN
550 LET B$="C'E' LA PAKOLA "+P$+"":
560 LET F(1)=1:RETURN
570 LET DT=5
580 IF P<16 THEN LET B$="NIENTE DA APRIRE":RETURN
590 IF F(5)=0 THEN LET B$="NON HAI LA CHIAVE":RETURN
600 PRINT:PRINT "DUAL'E' IL NUMERO DELLA CASSETTA":INPUT YN
610 IF NL<YN THEN LET B$="LA CHIAVE NON VA BENE":RETURN
620 LET B$="LA CASSETTA E' APERTA- ORA LA BORSA E' TUA!":LET F(6)=1
630 RETURN
640 LET DT=5
650 IF EP<P THEN LET B$="CHI DOVREI SEGUIRE?":RETURN
660 LET NP=FN A(20):GOSUB 950:LET P=NP
670 IF FN A(10)>8 THEN LET P=KP
680 IF FN A(10)>7 THEN LET B$="L'HAI PERDUTO DOPO UN PO'!":RETURN
690 LET EP=P
700 LET B$="L'HAI TENUTO SOTTO OCCHIO":RETURN
710 PRINT:PRINT "QUANTI MINUTI":INPUT DT
720 RETURN
730 PRINT:PRINT "DOVE VOGLIAMO INCONTRARCI":INPUT S$
740 PRINT:PRINT "A CHE ORA (ORE.MINUTI)"
750 INPUT U
760 IF P=CP AND TICU AND K<CH THEN LET F(2)=1
770 LET DT=5:RETURN
780 LET B$="QUI NON C'E' NULLA":LET DT=10
790 IF P=KP THEN LET B$="HAI TROVATO UNA CHIAVE":LET F(5)=1
800 RETURN
810 LET DT=0:LET B$="SONO LE ORE "+STR$ (H)+". "+STR$ (M):RETURN
```

580-640: SUBROUTINE "APRIRE"

650-700: SUBROUTINE "SEGUIRE"

710-720: SUBROUTINE "ASPETTARE"

730-770: SUBROUTINE "LASCIARE"

780-800: SUBROUTINE "CERCARE"

810: SUBROUTINE "ORE"



```

820 LET DT=5
830 IF U=0 THEN GOTO 860
840 PRINT:PRINT "IL LUOGO D'INCONTRO E'"
850 PRINT S$;" ALLE "U
860 GOSUB 1320:RETURN
870 LET DT=0:LET B$="PREGO?":RETURN
880 PRINT "TROPPO TARDI "":STOP
890 PRINT:PRINT "BENISSIMO. LA TUA MISSIONE E' STATA UN SUCCESSO!"
900 LET TL=(FH-H)*60-M
910 LET S=INT ((20/NM+TL/120)*50)
920 PRINT:PRINT "IL TUO VOTO COME SPIA"
930 PRINT "E' "S"
940 STOP
950 LET NY=INT ((NP-1)/5)
960 LET NX=NP-5*NY
970 LET DX=ABS (X-NX):LET DY=ABS (Y-NY)
980 LET X=NK:LET Y=NY
990 LET D=SQRT (DX*DX+DY*DY)
1000 LET DT=INT (5+D):RETURN
1010 DIM R$(20),V$(11),F(7)
1020 FOR I=1 TO 20:READ R$(I)
1030 NEXT I

```

Il simbolo  $\hat{\cdot}$  significa "elevato a potenza". Sui computer VIC 20, C64 e Spectrum ha questo aspetto:  $\hat{\cdot}$

```

1040 FOR I=1 TO 11:READ V$(I):NEXT I
1050 RETURN
1060 DATA "AEROPORTO","FERMATA DELL'AUTOBUS","PONTE","CANALE","CHIESA"
1070 DATA "PARCO","BAR","BANCA","CINEMA","ALBERGO"
1080 DATA "CASINO","PIAZZA PRINCIPALE","UFFICIO POSTALE","STAZIONE DI POLIZIA"
1090 DATA "FIEKA"
1100 DATA "STAZIONE","MUNICIPIO","AMBASCIATA","GIARDINI","CASTELLO"
1110 DATA "ANDARE","DIRE","ESAMINARE","LEGGERE","APRIRE","SEGUIRE"
1120 DATA "ASPETTARE","LASCIARE","CERCARE"
1130 DATA "ORE","AIUTO"
1140 DATA "BUDINO","ARINGA AFFUMICATA","KORLA","FOCCACCIA"
1145 DATA "PAROLE CROCIATE","CANGURO"
1150 LET H=FN A(2)+8:LET M=0
1155 LET FH=FN A(2)+14
1170 LET CH=FN A(2)+H
1180 CLS:PRINT
1190 LET NE=0:LET T$="IL CONTATTO E' QUI"
1200 LET MP=FN A(18):LET KP=FN A(18)
1210 LET EP=FN A(18):LET CP=FN A(18)
1220 FOR I=1 TO FN A(6)
1230 READ P$:NEXT I
1240 LET NL=FN A(900)+99
1250 PRINT "APPUNTAMENTO DI SPIE"
1260 PRINT "=====
1270 PRINT "RITIRA IL MESSAGGIO PRESSO"
1280 AR=MP:GOSUB 1360:PRINT R$(MP)
1290 PRINT "IL CONTATTO LO RITIRERA' PRESSO"
1300 AR=CP:GOSUB 1360:PRINT R$(CP)
1310 PRINT "ALLE ORE "CH;"."00"
1320 PRINT "L'ULTIMO VOLO PARTE"
1330 PRINT "ALLE ORE "FH;"."00"
1340 PRINT:PRINT "PREMI RETURN PER CONTINUARE"
1350 INPUT D$:RETURN
1355 REM ROUTINE DI CONCORDANZA DI GENERE E NUMERO
1360 IF AR=1 OR AR=10 OR AR=13 OR AR=18 THEN PRINT "L'";
1370 IF AR=3 OR AR=4 OR AR=6 OR AR=17 THEN PRINT "IL ";
1380 IF AR=7 OR AR=9 OR AR=11 OR AR=20 THEN PRINT "IL ";
1390 IF AR=2 OR AR=5 OR AR=8 OR AR=12 OR (AR>13 AND AR<17)
THEN PRINT "LA ";
1400 IF AR=19 THEN PRINT "I ";
1410 RETURN

```

Le linee 1010-1340 costituiscono la sezione di "inizializzazione" del programma.

**820-860: SUBROUTINE "AIUTO"**

**870: Subroutine "parola non riconosciuta".**

**880: Messaggio "tempo scaduto".**

**890-940: Visualizza il messaggio e il punteggio di una missione conclusa con successo. Il punteggio dipende dal numero di mosse effettuate e dal tempo rimasto.**

**950-1000: Subroutine che calcola la distanza percorsa e il tempo impiegato dallo spostamento del giocatore.**

**1010-1050: Subroutine per la lettura dei dati.**

**1060-1145: Linee contenenti i dati.**

**1150: Sceglie l'ora d'inizio.**

**1160: Decide l'ora di partenza dell'ultimo volo.**

**1170: Decide l'ora in cui il contatto raccoglierà il messaggio.**

**1200-1210: Sceglie le posizioni del messaggio, della chiave, del nemico e il posto dove il contatto troverà il messaggio.**

**1220-1230: Sceglie la parola d'ordine.**

**1240: Sceglie il numero della cassetta.**

**1250-1340: Visualizza la presentazione del gioco.**

## Linee di conversione

- 10 DEF FNA(X)=INT(RND\*X)+1
- 10 Tralascia questa linea
- 70,1180 Sostituisci CLS con HOME
- ▲ 70,1180 Sostituisci CLS con PRINT CHR\$(147)
- 235 IF LEN(I\$)<9 THEN LET I\$=I\$+" "GOTO 235
- 300 GOSUB 360\*(V=1)+420\*(V=2)+490\*(V=3)+540\*(V=4)+570\*(V=5)+640\*(V=6)+710\*(V=7)+730\*(V=8)+780\*(V=9)+810\*(V=10)+820\*(V=11)+870\*(V=12)
- 365 IF LEN(N\$)<19 THEN LET N\$=N\$+" "GOTO 365
- 735 IF LEN(S\$)<19 THEN LET S\$=S\$+" "GOTO 735
- 1010 DIM R\$(20,19):DIM V\$(11,9):DIM F(7)
- 280,340,660,670,680,1150,1160,1170,1200,1210,1220,1240 Sost. FNA() con RND() p.es.: FNA(20) diventerà RND(20)

Prova a studiare attentamente ogni subroutine per vedere che funzione hanno le varie linee di programma.

Le variabili F(numero) sono variabili "flag". F(5), per esempio, vale 1 se possiedi la chiave e 0 se non ce l'hai. Cerca le altre variabili "flag" e prova a capire a che cosa servono.

# Codice Morse

Se vuoi essere una spia veramente in gamba, devi essere in grado di mandare, ricevere, e ovviamente intercettare messaggi in codice Morse. Questo programma ti aiuterà a impararlo. Se non hai mai usato il codice Morse prima d'ora, dovrai farti una tabella delle lettere e dei corrispondenti segnali Morse. Usa le linee di programma dalla 400 alla 450 per fare la tabella: esse riportano i codici Morse di ogni lettera in ordine alfabetico.

## Che cosa devi fare

Nel codice Morse ogni lettera è rappresentata da una serie di suoni o segnali luminosi lunghi o brevi. Questo programma fa uso di un asterisco lampeggiante. Ti darà il codice di una lettera e poi ti chiederà qual'era la lettera. Dovrai osservare con molta attenzione per distinguere i segnali lunghi e brevi e ricordarteli. Vedrai lampeggiare anche il cursore sulla sinistra dello schermo: ignoralo, non ha niente a che fare col codice.

## Come funziona

10: Va alla subroutine di lettura dei dati.

30: Decide la velocità.

50-80: Ti chiede a che livello vuoi giocare e decide in base alla tua risposta la velocità dei segnali.

110: Pausa.

120: Va alla subroutine per la scelta casuale di una lettera; la immagazzina in Q\$.

130: Trova il codice della lettera in punti e linee e lo immagazzina in F\$.

140: Va alla subroutine che visualizza un asterisco lampeggiante.

160-170: Aspetta la tua risposta.

180-190: Controlla se la risposta è esatta e stampa un messaggio.

200-210: Si ferma e torna alla lettera successiva.

220-250: Controlla F\$ per tutta la sua lunghezza, ponendo K uguale a 1 per i punti e uguale a 3 per le linee.

260-270: Va alla subroutine che visualizza l'asterisco. Torna indietro, pone K uguale a 1 e poi torna alla subroutine della linea 340 per "stampare" uno spazio. Ciò darà l'intervallo tra punti e linee.

280: Ritorna al punto o alla linea successivi.

310-320: Subroutine per la scelta casuale di una lettera.

330: "Accende" l'asterisco.

340: Fa una pausa di durata dipendente da K.

350: Stampa uno spazio invece dell'asterisco.

370-390: Subroutine per la lettura dei dati.

400-450: Linee contenenti i dati.

Man mano che impari le lettere, prova a fare a meno della tabella.

La velocità è controllata dalla linea 30. Sui computer BBC ed Electron, cambia in 100 il valore di S. (Più piccolo è il numero, più veloce sarà il programma).

Potresti provare ad aggiungere una routine per il punteggiamento. (Il programma Robot-Spia potrebbe darti un'idea su come fare).

Prova aggiungere anche dei suoni, con la visualizzazione dell'asterisco, in sua vece. Ricorda: che un punto è un suono breve e una linea è tre volte più lungo.

```

3 REM CODICE MORSE
.5 REM VERSIONE GENERALE
10 GOSUB 370
●▲20 CLS
30 LET S=30
40 PRINT:PRINT "CODICE MORSE"
50 PRINT:PRINT "VELOCITA' ?"
60 PRINT:PRINT "(1=VELOCE)"
70 PRINT "(5=LENTO)"
80 INPUT P:LET P=P*S
●▲90 CLS
100 :PRINT:PRINT "PREPARATI"
110 FOR T=1 TO 20*S:NEXT T
120 GOSUB 310
□130 LET F$=M$(ASC(Q$)-64)
140 GOSUB 220
●▲150 CLS:PRINT
160 PRINT "SCRIVI LA TUA RISPOSTA"
170 INPUT X$
180 IF X$=Q$ THEN PRINT "ESATTO"
190 IF X$<>Q$ THEN PRINT "NO. LA RISPOSTA ERA: ";Q$
200 FOR T=1 TO 30*S:NEXT T
210 GOTO 90
220 FOR J=1 TO LEN(F$)
□230 LET W$=MID$(F$,J,1)
□240 IF W$="." THEN LET K=1
250 IF W$="-" THEN LET K=3
260 GOSUB 330:LET K=1
270 GOSUB 340
280 NEXT J
290 RETURN
300 PRINT
■□310 LET Q$=CHR$(INT(RND(1)*26+65))
320 RETURN
□●▲330 PRINT TAB(10,10);"*"
340 FOR T=1 TO P*K:NEXT T
□●▲350 PRINT TAB(10,10);" "
360 RETURN
□370 DIM M$(26)
380 FOR I=1 TO 26:READ M$(I):NEXT I
390 RETURN
400 DATA ".","-",".-","-.-","-.-.-","-.-.-.-"
410 DATA "-.-.-.-.-","-.-.-.-.-.-","-.-.-.-.-.-.-"
420 DATA "-.-.-.-.-.-.-.-","-.-.-.-.-.-.-.-.-"
430 DATA "-.-.-.-.-.-.-.-.-.-","-.-.-.-.-.-.-.-.-.-.-"
440 DATA "-.-.-.-.-.-.-.-.-.-.-.-","-.-.-.-.-.-.-.-.-.-.-.-.-"
450 DATA "-.-.-.-.-.-.-.-.-.-.-.-.-.-"

```

## Linee di conversione

```

● 20,90,150 Sostituisci CLS con HOME
▲ 20,90,150 Sostituisci CLS con PRINT CHR$(147)
130 LET F$=M$(CODE(Q$)-64)
□ 230 LET W$=F$(J)
235 IF W$="." THEN GOTO 280
310 LET Q$=CHR$(INT(RND*26+65))
■ 310 LET Q$=CHR$(RND(26)+64)
□ 330 PRINT AT 10,10;""
● 330 VTAB(10):PRINT TAB(10);""
▲ 330 PRINT CHR$(19):FOR I=1 TO 10:PRINT:NEXT:PRINT TAB(10);""
■ 330 PRINT @ 240;""
□ 350 PRINT AT 10,10;""
● 350 VTAB(10):PRINT TAB(10);""
▲ 350 PRINT CHR$(19):FOR I=1 TO 10:PRINT:NEXT:PRINT TAB(10);""
■ 350 PRINT @ 240;""
□ 370 DIM M$(26,4)

```



# Soluzioni dei quesiti

Ecco le soluzioni di alcuni dei quesiti di questo libro. Le tue soluzioni potrebbero essere differenti, ma ciò non ha importanza purché funzionino. Controlla però che siano chiare e semplici come queste riportate qui sotto.

## Occhi da Spia

Ecco come si fa per aggiungere una routine di punteggio massimo.

```
15 LET H=0
215 PRINT "IL PUNTEGGIO MASSIMO È";H
275 IF P>H THEN LET H=P:PRINT "HAI SUPERATO IL PRECEDENTE PUNTEGGIO
    MASSIMO"
290 INPUT A$:IF A$="S" THEN GOTO 20
```

## Faro nella notte

Ecco come fare per tenere acceso il faro più a lungo.

```
430 IF C1=TC THEN LET L=1:LET C1=0:LET TC=INT(RND(1)*10+(8-A))+10:
    GOSUB 300
```

... ed ecco come si dà una seconda possibilità al giocatore.

```
15 LET NP=0
255 LET NP=NP+1
257 IF NP<2 THEN FOR T=1 TO 200:NEXT T:GOTO 30
```



Aumenta questo numero per avere più di 2 possibilità.



Questo loop fa fare una pausa per permetterti di leggere il messaggio. Aumenta il numero del loop per ottenere una pausa più lunga.



Usa la versione RND del tuo computer.



NP è una nuova variabile che sta per "numero di partite".

## Robot-Spia

Con queste linee il computer chiederà il nome al giocatore se viene superato il precedente punteggio massimo.

```
25 LET H$="NESSUNO"
330 Togli questa linea
365 IF E=1 AND S>H THEN LET H=S:GOSUB 410
410 PRINT "HAI SUPERATO IL PRECEDENTE PUNTEGGIO MASSIMO!"
420 PRINT "IL TUO NOME, PREGO"
430 INPUT H$:RETURN
```

## Prova di Abilità

Ecco come si fa per aumentare il numero delle posizioni da 10 a 15.

```
10 DIM N(15):DIM N$(5)
70,160,200,460 IN QUESTE LINEE SOSTITUISCI IL 10 CON 15
```



# GIOCHI DEL BRIVIDO



## Indice

- 22 I programmi dei  
giochi del brivido
- 23 Incubo col computer
- 24 Il Mago dei numeri
- 26 Acchiappafantasm
- 28 La Donna Ragno
- 30 Lo Scavafosse
- 32 La Casa Folle
- 34 Il Labirinto dello  
Spettro
- 37 Soluzioni dei quesiti

## I programmi dei giochi del brivido

Questi programmi sono scritti per dei giochi semplici da giocare su un personal computer. Essi funzionano sui computer ZX81 (con espansione di memoria), ZX Spectrum, BBC, VIC 20, Dragon, Oric, TRS-80 e Apple. Il listato principale di ogni gioco funziona sullo ZX81, e le linee che devono essere convertite per i vari computer sono contrassegnate coi seguenti simboli:

- ▲ VIC e Pet
- ★ BBC
- TRS-80
- Apple
- ) ZX Spectrum
- ◇ Dragon
- Oric

Ogni volta che vedi il simbolo del computer che stai usando all'inizio di una linea, sostituiscila con la linea in calce al programma che porta il simbolo corrispondente.

### Come far girare i programmi

Quando hai introdotto il programma, controllalo con attenzione. Ricordati che è facile commettere errori quando si copiano i programmi, anche se sei molto esperto. Per cominciare il gioco, dai il RUN. Se il programma non funziona correttamente, è molto probabile che contenga ancora qualche errore, per cui fai il LIST del programma e controllalo di nuovo. Quando la partita finisce, il computer ti comunicherà qualcosa del

tipo BREAK IN 200. Per giocare ancora, devi dare di nuovo il RUN.

### Esperimenti coi programmi

Nel libro troverai dei suggerimenti su modifiche che potrai fare ai programmi, ma non devi aver timore di sperimentare cambiamenti inventati da te. Poiché i programmi sono stati scritti per funzionare su molti computer diversi, spesso non fanno uso di grafiche, colori o suoni. Potresti provare a riscrivere i programmi per il tuo computer, arricchendoli di grafiche, suoni e qualsiasi altro effetto che il tuo computer consenta.

### Come cambiare la velocità

In alcuni dei giochi troverai anche le istruzioni per modificarne la velocità. Potresti anche ritenerlo necessario, dato che molti di questi giochi dipendono sia dalla velocità del tuo computer che dalla tua prontezza di riflessi. Quasi certamente, dopo che avrai giocato un po' di volte, vorrai renderli comunque più veloci.

*Incubo col Computer è ispirato ad un'idea di  
Brendon Kavanagh*

*Il Mago dei Numeri, Acchiappafantasmi e Il  
Labirinto dello Spettro sono stati scritti da  
Colin Reynolds*

*La Donna Ragno è stato scritto da Val  
Robinson, Lo Scavafosse da Alan Ramsey e La  
Casa Folle da Keith Campbell.*



# Incubo con il Computer

Sei un maniaco del computer che veglia fino a tarda notte, e ti sei addormentato sulla tastiera. Improvvisamente, il tuo computer si anima e comincia a lanciare numeri e a maltrattarti. Per domarlo, devi riuscire a colpire i numeri, man mano che appaiono sullo schermo, premendo il tasto del numero corrispondente a quello "lanciato" dal computer. Il tuo punteggio iniziale (300 punti) cresce se premi il tasto giusto e diminuisce ogni volta che sbagli. Se riesci ad arrivare a 500 punti, il computer si arrenderà e vincerai tu, ma se il punteggio cala fino a zero, diventerai schiavo del tuo computer. È un vero e proprio incubo. Riuscirai a non impazzire?

## Come funziona

5 LET F\$=""	
10 DIM C\$(5,23)	
20 LET S=300	Pone il punteggio uguale al valore iniziale.
30 LET C\$(1)="* VIVA I COMPUTER! *	
40 LET C\$(2)="GLI UOMINI SONO STUPIDI"	Immagazzina in memoria i commenti.
50 LET C\$(3)="UN ROBOT PER PRESIDENTE!"	
60 LET C\$(4)="I COMPUTER SONO GRANDI!"	
70 LET C\$(5)="*SONO PIU' IN GAMBA IO! *	
80 CLS	Pulisce lo schermo.
90 LET N=INT(RND*9)+1	Sceglie un numero tra 1 e 9 e lo pone in N.
100 PRINT TAB(5);N	
110 PRINT TAB(15);S	Stampa N e il tuo punteggio (S) in punti differenti dello schermo.
120 IF RND>0.5 THEN GOTO 150	
130 PRINT	Decide se andare alla linea 150.
140 PRINT C\$(INT(S/100)+1)	
150 IF S<60 THEN PRINT"<NON HAI SPERANZE>"	Stampa messaggi diversi a seconda del tuo punteggio.
160 IF S<40 THEN PRINT"URCA! AIUTO!"	
170 FOR I=1 TO 10	
180 LET A\$=INKEY\$	Va a controllare se stai premendo un tasto, e in questo caso, lo immagazzina in F\$.
190 IF A\$<>" " THEN LET F\$=A\$	
200 NEXT I	Diminuisce il punteggio di 10.
210 LET S=S-10	
220 IF VAL(F\$)>N THEN GOTO 240	Se hai premuto il tasto sbagliato, il computer salta alla linea 240 per evitare l'aumento del punteggio della linea 230.
230 LET S=S+10+N*2	
240 IF S<0 THEN GOTO 270	Controlla il punteggio per vedere se hai vinto o perso e salta alla routine di chiusura.
250 IF S>500 THEN GOTO 290	
260 GOTO 80	Torna indietro per un altro gioco.
270 PRINT"ORA SEI MIO SCHIAVO"	
280 STOP	
290 PRINT"OK. HAI VINTO (PER STAVOLTA)"	
300 STOP	

Il listato in alto funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

```

* 10 DIM C$(5)
  80 HOME
  80 PRINT CHR$(147)
* 90 LET N=INT(RND(1)*9)+1
  90 LET N=INT(RND(0)*9)+1
* 120 IF RND(1)>0.5 THEN GOTO 150
  120 IF RND(0)>0.5 THEN GOTO 150
* 170 FOR I=1 TO 400
  175 A$=""
  180 IF PEEK(-16384)>127 THEN GET A$
  180 LET A$=INKEY$(0)
  180 GET A$
  180 LET A$=KEY$

```

La velocità del gioco dipende da questo numero. Probabilmente dovrai modificarlo per adattare la velocità del computer alla tua prontezza di riflessi. Un numero più alto rallenterà il gioco.

## Modifiche da provare

Le linee 20, 210 e 230 controllano il punteggio. Cambia i numeri in queste linee e guarda cosa succede.

## Quesito

Sei capace di modificare il programma in modo che visualizzi lettere invece che numeri?

## Il Mago dei Numeri



Questo è il gioco che il Mago dei Numeri propone a chi gli fa visita. Mediante un incantesimo ferma a mezz'aria i numeri da 1 a 9, e poi fa ruotare due dadi. Devi dargli allora due numeri tra quelli che appaiono a mezz'aria che, sommati, diano lo stesso totale dei numeri sui dadi. Una volta usato un numero, la sua immagine sparirà, e non potrai usarlo un'altra volta. Vinci se fai sparire tutti i numeri prima di usare tutte le possibilità. (Avrai una possibilità in più ogni volta che dal lancio del mago usciranno due numeri uguali). Puoi usare lo zero come numero quante volte vuoi. Se non riesci in un turno, vai al prossimo, introducendo lo zero due volte. Guarda quante volte riesci a battere il Mago.

## Come funziona


2 REM IL MAGO DEI NUMERI 3 REM VERSIONE GENERALE 10 PRINT "IL MAGO DEI NUMERI" 20 DIM A(10)	Prepara l'array A.
30 LET T=8	Stabilisce il numero di turni.
40 FOR I=1 TO 10 50 LET A(I)=1 60 NEXT I	Mette un 1 in ogni spazio di A.
▲●70 CLS 80 LET A(1)=1	Si assicura che A(1) contenga sempre 1 (ciò ti permette di usare lo zero quante volte vuoi).
90 LET V=0	V tiene conto di quanti numeri sono stati usati (all'inizio viene azzerato).
100 FOR I=2 TO 10 110 IF A(I)=0 THEN GOTO 150 120 PRINT " " 130 LET V=V+1 140 GOTO 160 150 PRINT I-1 160 NEXT I 170 PRINT 180 IF V=9 THEN GOTO 370	Visualizza i numeri. Se A(1)=0, il numero è già stato usato, per cui lascia uno spazio e aggiunge 1 a V.
190 PRINT "HAI ANCORA "T;" POSSIBILITÀ" ■▲●200 LET C=INT(RND*6+1) ■○●210 LET B=INT(RND*6+1)	Ti dice quante possibilità ti restano ancora.
220 PRINT "IL LANCIO HA DATO "C;" / "B 230 PRINT "QUALI SONO I TUOI NUMERI?" 240 IF B=C THEN LET T=T+2	Sceglie i numeri per il lancio dei dadi e li visualizza.
250 LET T=T-1 260 IF T<0 THEN GOTO 350	Se sui due dati è uscito lo stesso numero, ti dà una possibilità in più.
270 INPUT N 280 INPUT M 290 IF M>9 OR N>9 THEN PRINT "TROPPO ALTO. PROVA DI NUOVO" 300 IF M>9 OR N>9 THEN GOTO 230	Diminuisce le possibilità di 1; controlla che T non sia zero.
310 IF M+N>B+C OR A(N+1)=0 OR A(M+1)=0 THEN GOTO 70	Ti chiede due numeri e controlla che siano validi.
320 LET A(M+1)=0 330 LET A(N+1)=0	Se non danno il totale giusto, o uno di essi è già stato usato, torna indietro per il giro successivo.
340 GOTO 70 350 PRINT "HA VINTO IL MAGO" 360 STOP 370 PRINT "HAI VINTO TU" 380 STOP	Se i numeri sono giusti, gli 1 nei relativi spazi in A vengono sostituiti da altrettanti 0, per mostrare che sono già stati usati.

Il listato in alto funziona sullo ZX81.  
Per gli altri computer, dovrai fare queste modifiche.

●70 HOME  
▲70 PRINT CHR\$(147)  
■○●200 LET C=INT(RND(1)\*6+1)  
■●200 LET C=INT(RND(0)\*6+1)  
■○●210 LET B=INT(RND(1)\*6+1)  
■●210 LET B=INT(RND(0)\*6+1)

Prova a fare in modo che il computer torni automaticamente indietro per un'altra partita, quando una finisce, e che registri anche quante volte vinci e quante volte perdi. Potresti anche inventare una punizione terribile per i giocatori che perdono troppo....

## Acchiappafantasma



I fantasmi in questo gioco sono numeri che svolazzano sullo schermo. Per acchiapparli, devi mettere in moto il tuo aspiratore di fantasmi, premendo il tasto X, ma esso funzionerà solo se il suo numero è lo stesso del fantasma che ti sta attaccando. Puoi aumentare il numero dell'aspiratore, premendo il tasto M (dopo che è arrivato al 9, torna di nuovo allo zero). Se non riesci ad aspirare un fantasma, esso ti priverà di una delle vite (indicate come / alla sommità dello schermo). Vediamo se sei bravo ad acchiappare fantasmi!

```
3 REM ACCHIAPPAFANTASMI
5 REM VERSIONE GENERALE
10 PRINT "ACCHIAPPAFANTASMI"
20 LET S=0
30 LET Y=0
```

### Come funziona

Azzera il punteggio e il tuo numero.

```
40 LET L=3
```

Pone il numero di "vite" uguale a 3.

```
50 LET N=INT (RND*10)
```

Sceglie a caso un numero tra 0 e 9 per il fantasma.

```
60 LET I=1
```

Fa partire il contatore per la distanza che il fantasma può percorrere prima di raggiungerti.

```
70 CLS
```

```
80 FOR J=1 TO L
```

```
90 PRINT "/"
```

```
100 NEXT J
```

```
110 PRINT
```

```
120 PRINT TAB(I);N;TAB(18);"/";Y
```

Loop per visualizzare le tue "vite", come / nell'angolo in alto a sinistra dello schermo.

Visualizza alle posizioni esatte sullo schermo il numero del fantasma, la barriera ":" e te(y).

Questo è un gioco adatto per aggiungerci suoni. Prova ad aggiungere delle subroutine per qualcosa che faccia gelare il sangue quando il giocatore perde una "vita", e qualche altro suono per quando i fantasmi vengono aspirati.





```
130 LET B$=INKEY$
140 IF B$="X" THEN GOTO 220
150 IF B$="M" THEN LET Y=Y+1

160 IF Y=10 THEN LET Y=0
170 LET I=I+1
180 IF I<18 THEN GOTO 70
190 LET L=L-1
200 IF L>0 THEN GOTO 50
210 GOTO 270
220 IF Y<N THEN GOTO 170
230 PRINT "PRESO"
240 PRINT "*****"
250 LET S=S+(18-I)
260 GOTO 50
270 CLS
280 PRINT "IL TUO PUNTEGGIO COME"
290 PRINT "ACCHIAPPAFANTASMI E'":S
300 PRINT
310 PRINT "UN'ALTRA PARTITA?"
320 INPUT A$
330 IF A$="S" THEN RUN
340 IF A$="N" THEN STOP
```

Puoi modificare la velocità di questo gioco, cambiando il numero alla linea 125. Per renderlo più veloce, usa un numero più piccolo. Puoi renderlo più difficile, riducendo i numeri alla linea 120.



Controlla se stai premendo un tasto.

Se premi X (attivi l'acchiappafantasmi) il computer va a fine programma per vedere se puoi farlo. Se premi M, aumenta il numero di 1.

Azzerà il tuo numero quando ha raggiunto 10.

Aumenta il contatore della distanza del fantasma; se non ha raggiunto la barriera e tu non l'hai aspirato, il computer torna a stamparlo nella posizione successiva.

Se ha raggiunto la barriera, perdi una vita.

Torna indietro per un altro fantasma se ti restano ancora vite.

Controlla se il tuo numero era come quello del fantasma quando l'hai aspirato; se non lo era torna indietro per il fantasma successivo.

Il listato in alto funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

```
150 LET N=INT(RND(1)*10)
50 LET N=INT(RND(0)*10)
70,270 HOME
70,270 PRINT CHR$(147)
120 PRINT SPC(I);N;SPC
(18-I);": ";Y
125 FOR K=1 TO 50
127 NEXT K
128 LET B$=""
130 IF PEEK(-16384)>127
THEN GET B$
130 LET B$=INKEY$(0)
130 GET B$
130 LET B$=KEY$
245 FOR K=1 TO 300
247 NEXT K
```



# La Donna Ragno

**Aiuto, la Donna Ragno!**


**Devi essere fortunato, oggi è di buon umore. Se riesci a indovinare la lettera che sta pensando, e abbastanza velocemente, ti lascerà andare.**

**Altrimenti, probabilmente verrai trasformato in mosca.**

**Per scoprire quale sia la lettera, dai una parola alla Donna Ragno, ed essa ti dirà se la parola contiene o no quella lettera.**



## Come funziona



Il listato in alto funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

```

● 20 HOME
▲ 20 PRINT CHR$(147)
) 30 LET T=INT(RND*26+65)
○ 30 LET T=INT(RND(1)*26+65)
◇ 30 LET T=INT(RND(0)*26+65)
■ 170 LET A$=MID$(W$, I, 1)
* 230 FOR A=1 TO 500
● 260 HOME
▲ 260 PRINT CHR$(147)
  
```

## Esperimenti da provare

Guarda cosa succede se sostituisci il 15 alla linea 250 con un numero più piccolo. E se cambi il 4 e l'8 alla linea 140?

## Come modificare il gioco

La Donna Ragno ne ha abbastanza, e vuole che tu modifichi il programma nel modo seguente:

Se la tua parola contiene la lettera pensata, ti permette due possibilità per tirare a indovinare, ma se sbagli, perdi 5 tentativi. Sei capace di farlo?

# Lo Scavafosse

È una notte buia e tempestosa: non è di certo adatta per perdersi in un cimitero, ma a te è capitato. Hai tempo fino a mezzanotte per trovare l'uscita. Scheletri si acquattano nell'ombra, aspettando che ti avvicini a loro per spaventarti a morte. Puoi scavare delle fosse per tenerli lontani, ma scavare è un lavoro faticoso per cui non puoi fare più di cinque fosse per partita. Devi inoltre stare attento a non cadere tu stesso nelle fosse che hai scavato. Le lapidi (simbolo +) e le mura di recinzione del cimitero (simbolo :) ti bloccano la strada. Le fosse che scavi sono indicate con 0, tu sei l'asterisco e gli scheletri sono le X. Riuscirai a sfuggire?

## Come funziona

50-90	Riempie l'array A con spazi vuoti.
110	Stabilisce il numero massimo permesso di fosse.
120-170	Codifica i simboli da usare.
180-260	Mette i muri nell'array.
270-310	Sceglie a caso le posizioni per le lapidi.
320-390	Stabilisce le posizioni iniziali del giocatore e degli scheletri.
400-420	Mette gli scheletri nell'array.
440-460	Calcola la tua nuova posizione.
650-740	Controlla cosa c'è nella nuova posizione. Se non c'è nulla, effettua lo spostamento.
770-850	Visualizza messaggi diversi a seconda dell'esito della partita.
860-980	Visualizza il cimitero.
1030-1100	Mette una fossa nell'array se la vuoi.
1110-1310	Muove gli scheletri.

```
3 REM LO SCAVAFOSSE
5 REM VERSIONE GENERALE
▲●10 CLS
20 PRINT "SCAVAFOSSE"
30 DIM A(10,20)
40 DIM B(6)
50 FOR I=1 TO 10
60 FOR J=1 TO 20
■○●▲70 LET A(I,J)=CODE(" ")
80 NEXT J
90 NEXT I
100 LET M=0
110 LET X=5
120 LET Y=CODE("*")
130 LET B=CODE("+")
○■▲140 LET C=CODE("0")
■○150 LET D=CODE(":")
160 LET E=CODE("X")
170 LET Z=CODE(" ")
180 FOR J=1 TO 10
190 LET A(J,1)=D
200 LET A(J,20)=D
210 NEXT J
220 FOR J=1 TO 20
230 LET A(1,J)=D
240 LET A(10,J)=D
250 NEXT J
260 LET A(9,20)=Z
270 FOR J=1 TO 20
○■▲280 LET F=INT(RND*7+2)
■○290 LET G=INT(RND*15+3)
300 LET A(F,G)=B
310 NEXT J
320 LET M=2
330 LET N=2
340 LET B(2)=19
350 LET B(4)=19
360 LET B(6)=19
370 LET B(1)=4
380 LET B(3)=3
390 LET B(5)=2
400 FOR J=1 TO 5 STEP 2
410 LET A(B(J),B(J+1))=E
420 NEXT J
430 GOSUB 860
440 PRINT
450 PRINT "MOSSA ";M
460 PRINT "(PUOI ANDARE A N,S,E,O"
470 INPUT A$
480 IF A$="N" THEN GOTO 530
490 IF A$="E" THEN GOTO 560
500 IF A$="S" THEN GOTO 600
510 IF A$="O" THEN GOTO 630
520 GOTO 470
530 LET T=N-1
```



```

540 LET U=M
550 GOTO 650
560 LET T=N
570 LET U=M+1
580 IF A(T,U)=Z AND M=19 THEN GOTO 770
590 GOTO 650
600 LET T=N+1
610 LET U=M
620 GOTO 650
630 LET T=N
640 LET U=M-1
650 IF A(T,U)=Z THEN GOTO 710
660 IF A(T,U)=D OR A(T,U)=B THEN GOTO 690
670 IF A(T,U)=C THEN GOTO 800
680 IF A(T,U)=E THEN GOTO 830
690 PRINT "LA STRADA E' BLOCCATA"
700 GOTO 440
710 LET A(N,M)=Z
720 LET N=T
730 LET M=U
740 LET A(N,M)=Y
750 GOSUB 1030
760 GOTO 400
770 PRINT "SEI LIBERO**"
780 PRINT "IL TUO RISULTATO E' ":INT((60-M)/60*(96+X)):"Z"
790 STOP
800 PRINT "SEI CADUTO IN"
810 PRINT "UNA DELLE TUE BUCHE"
820 STOP
830 PRINT "SEI STATO SPAVENTATO"
840 PRINT "A MORTE DA UNO SCHELETRO"
850 STOP
860 LET A(N,M)=Y
▲●870 CLS
880 LET R=N
890 LET S=M
900 LET W=N+1
910 IF W>60 THEN GOTO 990
920 FOR I=1 TO 10
930 FOR J=1 TO 20
940 PRINT CHR$(A(I,J))
950 NEXT J
960 PRINT
970 NEXT I
980 RETURN
990 PRINT "L'OROLOGIO HA SUONATO"
1000 PRINT "LA MEZZANOTTE"
1010 PRINT "AAAAHHHHH!!"
1020 STOP
1030 IF X=0 THEN GOTO 1110
1040 PRINT
1050 PRINT "VUOI SCAVARE"
1060 PRINT "UNA FOSSA (S O N)"
1070 INPUT B$
1080 IF B#<>"S" THEN GOTO 1110

```

```

1090 LET X=X-1
1100 LET A(R,S)=C
1110 FOR J=1 TO 5 STEP 2
1120 LET T=B(J)
1130 LET U=B(J+1)
1140 IF A(T,U+1)=Y THEN GOTO 830
1150 IF A(T,U-1)=Y THEN GOTO 830
1160 IF A(T-1,U)=Y THEN GOTO 830
1170 IF A(T+1,U)=Y THEN GOTO 830
1180 IF A$="O" THEN GOTO 1300
1190 IF A$="S" AND A(T+1,U)=Z THEN
    GOTO 1270
1200 IF A$="N" AND A(T-1,U)=Z THEN
    GOTO 1280
1210 IF A$="E" AND A(T,U+1)=Z AND
    M>U THEN GOTO 1240
1220 IF A$="E" AND A(T,U-1)=Z
    THEN GOTO 1250
1230 GOTO 1300
1240 LET B(J+1)=B(J+1)+2
1250 LET B(J+1)=B(J+1)-1
1260 GOTO 1290
1270 LET B(J)=B(J)+2
1280 LET B(J)=B(J)-1
1290 LET A(T,U)=Z
1300 NEXT J
1310 RETURN

```

**Il listato in alto funziona sullo ZX81.  
Per gli altri computer, dovrai fare  
queste modifiche.**

- 10, 870 HOME
- ▲ 10, 870 PRINT CHR\$(147)
- ◻ ○ ◐ ◑ ▲ ● 70, 120-170 MODIFICA CODE IN ASC
- ◻ ○ ▲ ● 280, 290 MODIFICA RND IN RND(1)
- ◐ ◑ 280, 290 MODIFICA RND IN RND(0)





# La Casa Folle

Ti trovi intrappolato in una casa dove tutto si muove, comprese le pareti. Se le porte si trovano allineate, anche per un momento solo, potrai fare uno scatto verso la libertà. Hai trovato un quadro comandi che sembra essere in grado di controllare alcuni dei movimenti della casa. I tasti X e C fanno cambiare direzione alla porta più vicina a te (in alto sullo schermo); i tasti N e M hanno lo stesso effetto sulla porta più distante da te, ma sembra che non ci sia alcuna possibilità di controllare la porta centrale. Mentre stai freneticamente lottando con i comandi del quadro, senti dei passi nel corridoio dietro di te. Il numero nell'angolo in alto a sinistra sullo schermo mostra l'avvicinarsi di quei passi. Se non riesci a scappare prima che i passi-fantasma ti raggiungano ...

... aaggghhhh!

Se il tuo computer ha il suono, prova a creare degli effetti sonori per i passi e per la fine del gioco, nel caso in cui il giocatore non riesca a scappare

## Come funziona

Prepara due array. Il primo si riferisce alla posizione (P) di ciascuna uscita e l'altro alla direzione (S) in cui essa si sposta.

Crea una fila di asterischi di lunghezza pari alla larghezza dello schermo (W) e la mette in L\$.

Pone il contatore dei passi al valore iniziale.

Stabilisce la direzione del movimento delle porte all'inizio del gioco.

Sceglie casualmente le posizioni iniziali delle tre uscite.

Visualizza pareti e porte.

Visualizza il contatore dei passi in alto a sinistra sullo schermo.

Controlla se i passi ti hanno raggiunto, e in questo caso va alla fine del gioco.

Controlla se hai vinto.

Controlla la tastiera per le istruzioni. Se necessario, cambia direzione al movimento modificando i segni + e -.

Sposta l'uscita centrale ogni 25 passi.

Diminuisce il valore del contatore dei passi.

Trova la nuova posizione delle porte. Se la porta raggiunge il bordo dello schermo, il computer la tiene ferma in quel punto.

Torna alla mossa successiva.

Visualizza la tua fuga attraverso le uscite con un messaggio.

Stampa il messaggio di sconfitta.

```

10 DIM P(3)
20 DIM S(3)
30 CLS
40 PRINT "LA CASA FOLLE"
50 LET L$=""
60 LET W=31
70 FOR I=1 TO W
80 LET L$=L$+"*"
90 NEXT I
100 LET CT=240
110 LET S(1)=1
120 LET S(3)=-1
130 FOR I=1 TO 3
140 LET P(I)=INT(RND*(W-4)+1)
150 NEXT I
160 CLS
170 FOR I=5 TO 15 STEP 5
180 PRINT AT I,0;L$
190 PRINT AT I,P(I/5);">"
200 NEXT I
210 PRINT AT 0,0;CT
220 IF CT<0 THEN GOTO 450
230 IF P(1)=P(2) AND P(2)=P(3) THEN GOTO 390
240 LET Z$=INKEY$
250 IF Z$="" THEN GOTO 300
260 IF Z$="X" THEN LET S(1)=-1
270 IF Z$="C" THEN LET S(1)=1
280 IF Z$="N" THEN LET S(3)=-1
290 IF Z$="M" THEN LET S(3)=1
300 IF CT/25=INT(CT/25) THEN LET P(2)=P(2)+INT(RND(1)*20)-10
310 LET CT=CT-1
320 LET P(1)=P(1)+S(1)
330 LET P(3)=P(3)+S(3)
340 FOR I=1 TO 3
350 IF P(I)<1 THEN LET P(I)=1
360 IF P(I)>(W-4) THEN LET P(I)=W-4
370 NEXT I
380 GOTO 170
390 LET L=P(1)+1
400 FOR I=1 TO 15
410 PRINT AT I,L;"M"
420 NEXT I
430 PRINT "SEI LIBERO!!"
440 STOP
450 PRINT "TROPPO TARDI! I PASSI SI SONO FERMATI."
460 PRINT "ARGHHH!!!"
470 STOP

```



Il listato a fianco funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

○ ORIC

```
140 LET P(I)=INT(RND(1)*(W-4))+1
180 PLOT 1,I,L$
190 PLOT P(I/5),I,"> <"
210 PLOT 0,0,STR$(CT)
240 LET Z$=KEY$
300 IF CT/25=INT(CT/25) THEN LET
410 PLOT L,I,"M"
      P(2)=P(2)+INT(RND(1)*20)-10
```

★ BBC

```
60 LET W=40
140 LET P(I)=INT(RND(1)*(W-4))+1
180 PRINT TAB(0,I);L$
190 PRINT TAB(P(I/5),I);"> <"
210 PRINT TAB(0,0);CT
240 LET Z$=INKEY$(10)
300 IF CT/25=INT(CT/25) THEN LET
      P(2)=P(2)+INT(RND(1)*20)-10
410 PRINT TAB(L,I);"M"
```

▲ VIC-20

```
30 PRINT CHR$(147)
60 LET W=21
140 LET P(I)=INT(RND(1)*(W-4))+1
160 PRINT CHR$(147)
170 FOR I=1 TO 3
172 FOR J=1 TO 5
174 PRINT
180 NEXT J
190 PRINT LEFT$(L$,P(I));"> <";
      RIGHT$(L$, (18-P(I)))
210 PRINT CHR$(19);CT
240 GET Z$
300 IF CT/25=INT(CT/25) THEN LET
      P(2)=P(2)+INT(RND(1)*20)-10
375 PRINT CHR$(19)
395 PRINT CHR$(19)
410 PRINT TAB(L)"M"
```

● APPLE

```
30 HOME
140 LET P(I)=INT(RND(1)*(W-4))+1
160 HOME
180 VTAB(I)
185 PRINT L$
190 VTAB(I)
195 HTAB(P(I/5))
197 PRINT "> <"
210 VTAB(1):HTAB(1):PRINT CT
240 Z$=""
245 IF PEEK(-16384)>127 THEN GET Z$
300 IF CT/25=INT(CT/25) THEN LET
      P(2)=P(2)+INT(RND(1)*20)-10
395 VTAB(1)
410 PRINT TAB(L)"M"
```

■ TRS-80

```
60 LET W=63
140 LET P(I)=RND(60)
170 FOR I=2 TO 12 STEP 5
180 PRINT@ (I*64),L$
190 PRINT@ (I*64;P((I+3)/5)), "> <"
210 PRINT@0,CT
300 IF CT/25=INT(CT/25) THEN LET
      P(2)=P(2)+RND(20)-10
410 PRINT@ (I*64+L), "M"
```

◇ DRAGON

```
50,70,80,90 DELETE
140 LET P(I)=RND(28)
170 FOR I=2 TO 12 STEP 5
180 PRINT@32*I,STRING$(P((I+3)/5)-1,
      "*");"> <";STRING$(28-P
      ((I+3)/5), "*");
190 DELETE
210 PRINT@0,CT
300 IF CT/25=INT(CT/25) THEN LET
      P(2)=P(2)+RND(20)-10
410 PRINT@ (I*32+L), "M"
```

Che cosa succede se cambi il numero alla linea 100?  
E se togli il segno meno alla linea 120?  
Cambia i due 25 alla linea 300 con i numeri diversi e osserva la differenza.

# Il labirinto dello Spettro

È un posto lugubre. Gli scuri corridoi, tutti uguali tra loro, sembrano non portare da nessuna parte. Potrebbero essere anche infestati da spettri ... Puoi vedere soltanto dritto davanti a te e ti puoi muovere solo nella direzione verso cui stai guardando. Puoi girarti a destra e a sinistra, ma ciò servirà soltanto a mostrarti il labirinto da un'altra angolazione. Tutto ciò che devi fare è trovare la crocetta che segnala l'uscita. La tua posizione è indicata con una T e le pareti con dei segni #. Gulp! È veramente infestato dagli spettri! Gli spettri sono indicati dalle lettere S. Se te ne ritrovi uno proprio di fronte, verrai immediatamente trasportato in un altro punto del labirinto, senza sapere dove ti trovi e da che parte stai guardando. Ecco i tasti che puoi usare: X per spostarti in avanti, N per girarti a sinistra (di 90°), M per girarti a destra (di 90°).

10 DIM E(70)

20 DIM W\$(4,3)

30 DIM F(3)

40 LET W\$=""

50 LET W\$=W\$+"0000000000"

60 LET W\$=W\$+"0111100110"

70 LET W\$=W\$+"0010011100"

80 LET W\$=W\$+"0011010110"

90 LET W\$=W\$+"0110100100"

100 LET W\$=W\$+"0011111100"

110 LET W\$=W\$+"0000000000"

120 FOR I=1 TO 70

130 LET E(I)=VAL (W\$ (I TO I))

140 NEXT I

150 LET S=-1

160 LET G=0

170 LET X=INT (RND\*50)+10

180 IF E(X) < 1 THEN GOTO 170

190 GOSUB 860

200 LET D=INT (RND\*4)+1

210 IF X=G+10 OR Y=G-10 THEN GOTO 170

220 IF X=G+1 OR X=G-1 THEN GOTO 170

230 LET H=H+1

240 IF H=5 THEN GOSUB 860

250 GOSUB 390

260 LET A\$=INKEY\$

270 IF A\$="" THEN GOTO 260

280 IF A\$="M" THEN LET D=D+1

290 IF A\$="N" THEN LET D=D-1

300 IF D=5 THEN LET D=1

310 IF D=0 THEN LET D=4

320 IF A\$ < "X" THEN GOTO 210

330 IF D=1 AND E(X-10) < 0 THEN LET X=X-10

340 IF D=3 AND E(X+10) < 0 THEN LET X=X+10

350 IF D=2 AND E(X+1) < 0 THEN LET X=X+1

360 IF D=4 AND E(X-1) < 0 THEN LET X=X-1

370 IF E(X)=9 THEN GOTO 930

380 GOTO 210

Prepara un'area di immagazzinamento (array) per il labirinto. Prepara un mini-reticolo all'interno del labirinto che sarà la parte di esso che potrai vedere sullo schermo. Spazio di immagazzinamento per i dati per riempire una fila di questo mini-reticolo.

Contiene i dati per il labirinto: 1 = corridoio, 0 = muro, 9 = uscita.

Loop che mette i dati nell'array E. Per iniziare, pone uguale a -1 il numero di volte che lo spettro si è mosso.

Azzera il numero degli spettri, per iniziare.

Sceglie a caso la tua posizione di partenza.

Se questa posizione non si trova in un corridoio, torna a prenderne un'altra.

Va alla subroutine per il posizionamento dello spettro.

Per iniziare il gioco, sceglie a caso la direzione in cui ti trovi rivolto.

Controlla se ti trovi vicino a uno spettro e in questo caso ti sposta altrove.

Aumenta il numero di mosse che hai fatto.

Va alla subroutine per spostare lo spettro ogni 5 mosse.

Va alla subroutine che visualizza la sezione di labirinto che puoi vedere.

Ti chiede un'istruzione. Cambia il valore della direzione D, se occorre.

Torna alla mossa successiva se non ti sei mosso in avanti.

Controlla se ti puoi muovere.

Controlla se hai raggiunto l'uscita.

Torna alla mossa successiva se non l'hai raggiunta.

## Aggiungi rumori

Se il tuo computer ha il suono, aggiungi un rumore ogni volta che appare lo spettro.

```

390 FOR I=1 TO 4
400 LET T=I-1
410 GOTO 380+40*I
420 LET F(1)=X-10*T+1
430 LET F(2)=X-10*T
440 LET F(3)=X-10*T-1
450 GOTO 570
460 LET F(1)=X+10*T
470 LET F(2)=X+T
480 LET F(3)=X-10+T
490 GOTO 570
500 LET F(1)=X+10*T-1
510 LET F(2)=X+10*T
520 LET F(3)=X+10*T+1
530 GOTO 570
540 LET F(1)=X-10-T
550 LET F(2)=X-T
560 LET F(3)=X+10-T
570 FOR J=1 TO 3
580 IF F(J)<1 OR F(J)>69 THEN GOTO 840
590 IF E(F(J))=0 THEN LET V$(I,J)="#"
600 IF E(F(J))=1 THEN LET V$(I,J)=""
610 IF E(F(J))=9 THEN LET V$(I,J)="#"
620 IF E(F(J))=2 THEN LET V$(I,J)="S"
630 NEXT J
640 NEXT I
650 LET V$(1,2)="T"

```

```

660 CLS
670 PRINT
680 PRINT "LABIRINTO DELLO SPETTRO"
690 PRINT
700 PRINT "AVANTI      X"
710 PRINT "A DESTRA    M"
720 PRINT "A SINISTRA  N"
730 FOR I=1 TO 5
740 PRINT
750 NEXT I
760 FOR I=4 TO 1 STEP -1
770 LET P$=""
780 FOR J=3 TO 1 STEP -1
790 LET P$=P$+V$(I,J)
800 NEXT J
810 PRINT TAB(7);P$
820 NEXT I
830 RETURN
840 LET V$(I,J)="#"
850 GOTO 630
860 LET E(G)=1
870 LET G=INT(RND*50)+10
880 IF E(G)<1 THEN GOTO 870
890 LET E(G)=2
900 LET H=0
910 LET S=S+1
920 RETURN

```

```

930 PRINT "SEI RIUSCITO A EVADERE"
940 PRINT "IN ";S*5+H;" MOSSE"
950 STOP

```

Il listato a fianco funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

```

5 CLEAR
130 LET E(I)=VAL(MID$(W$,I,1))
170 LET X=INT(RND(1)*50)+10
170 LET X=INT(RND(0)*50)+10
200 LET D=INT(RND(1)*4)+1
200 LET D=INT(RND(0)*4)+1
255 LET A$=""
260 IF PEEK(-16384)>127 THEN GET A$
260 LET A$=INKEY$(0)
260 GET A$
260 LET A$=KEY$
410 ON D GOTO 420,460,500,540
660 HOME
660 PRINT CHR$(147)
870 LET G=INT(RND(1)*50)+10
870 LET G=INT(RND(0)*50)+10

```

## Migliora i simboli

Prova a sostituire i simboli +, G, T e # con caratteri grafici (e anche colori). Questo implicherà delle modifiche alle linee dalla 590 alla 620 e 840.

Subroutine che calcola:

- quali quadretti puoi vedere dal punto in cui ti trovi
- che cosa mettere in questi quadretti.
- visualizza le istruzioni e la sezione di reticolo che puoi vedere. "

## Ingrandisci il labirinto

Il labirinto è basato su una griglia contenente 70 quadretti. Per ingrandirlo dovrai modificare le seguenti linee, sostituendo N col numero di quadretti che vuoi che abbia il tuo nuovo reticolo. (N deve essere un multiplo di 10.)

- 10: Sostituisci 70 col tuo numero, N.
- 50-110: Aggiungi altre righe di numeri 1 e 0, con numeri di linea 55, 65 eccetera, in modo che il numero totale di caratteri del reticolo sia N.
- 120: Sostituisci 70 con N.
- 170-180: Sostituisci 50 con N-20.
- 580: Sostituisci 69 con N-1.

Subroutine per spostare lo spettro ogni cinque mosse.

Messaggio di fine partita.

# Seduta Spiritica

Stanno arrivando, lettera per lettera, dei messaggi dagli Spiriti. Essi vogliono che tu ti ricordi le lettere e che le digiti sulla tastiera del computer nella sequenza esatta. Se sbagli, si arrabbieranno, e molto...

Cerca asterischi sul tuo schermo: essi indicano le lettere nei messaggi degli Spiriti.

## Come funziona il programma

Dopo che hai copiato il programma e hai giocato un po' di volte, esamina le linee di programma e cerca a che cosa servono. (Suggerimento: D è una variabile che dice al computer su quale lato del quadrato -1, 2, 3 o 4 - si trova la successiva lettera).

```

5 REM SEDUTA SPIRITICA
7 REM VERSIONE GENERALE
10 LET S=0
20 LET G=0
30 LET CS=37
40 CLS
50 PRINT
60 PRINT "SEDUTA SPIRITICA"
70 FOR I=1 TO 3
80 LET X=6+I
90 LET Y=5
100 LET A$=CHR$(CS+I)
110 GOSUB 710
120 LET Y=Y+1
130 LET A$=CHR$(CS+22-I)
140 GOSUB 710
150 NEXT I
160 FOR I=1 TO 5
170 LET X=X+5
180 LET Y=Y+I
190 LET A$=CHR$(CS+27-I)
200 GOSUB 710
210 LET X=X+16
220 LET A$=CHR$(CS+8+I)
230 GOSUB 710
240 NEXT I
250 LET P$=""
260 LET N=INT(RND*4+3)
270 FOR I=1 TO N
280 LET A$="*"
290 LET L=INT(RND*26+1)
300 LET S$=CHR$(CS+L)
310 LET P$=P$+S$
320 LET D=4
330 IF L<22 THEN LET D=3
340 IF L<14 THEN LET D=2
350 IF L<9 THEN LET D=1
360 GOTO 340+30*D
370 LET Y=Y+6
380 LET X=X+6
390 GOTO 480
400 LET X=X+15
410 LET Y=Y-L-3
420 GOTO 480
430 LET Y=Y+10
440 LET X=X+28-L
450 GOTO 480
460 LET X=X+6
470 LET Y=Y+32-L
480 GOSUB 710
490 FOR T=1 TO 20
500 NEXT T
510 LET A$=" "

```

La linea 490 controlla la velocità alla quale appaiono gli asterischi. Per rendere il gioco più veloce, sostituisci il numero con uno più piccolo.

```

530 GOSUB 710
530 NEXT I
540 LET A$=""
550 LET X=0
560 LET Y=10
570 GOSUB 710
580 INPUT R$
590 IF R$=P$ THEN GOTO 670
600 LET G=G+1
610 IF G=1 THEN PRINT "IL
TAVOLO COMINCIA A MUOVERSI"
620 IF G=2 THEN PRINT "LE
LAMPADINE VANNO IN FRANTUMI"
630 IF G=3 THEN GOTO 730
640 FOR T=1 TO 100
650 NEXT T
660 GOTO 40
670 LET S=S+N
680 IF S<50 THEN GOTO 40
690 PRINT "GLI SPIRITI SE NE SONO ANDATI"
700 STOP
710 PRINT AT Y,X:A$
720 RETURN
730 PRINT "DUE MANI VISCIDE
TI AFFERRANO IL COLLO"
740 STOP

```

Il listato in alto funziona sullo ZX81. Per gli altri computer, dovrai fare queste modifiche.

```

30 LET CS=64
40 HOME
40 PRINT CHR$(147)
260 LET N=INT(RND(1)*4+3)
260 LET N=INT(RND(0)*4+3)
290 LET L=INT(RND(1)*26+1)
290 LET L=INT(RND(0)*26+1)
360 ON D GOTO 370,400,430,460
490 FOR T=1 TO 100
490 FOR T=1 TO 600
640 FOR T=1 TO 300
640 FOR T=1 TO 1000
710 PRINT TAB(X,Y):A$:
710 VTB(Y):HTAB(X+1):PRINT A$
710 PRINT CHR$(19):FOR K=1 TO Y:PRINT:
NEXT:PRINT TAB(X):A$
710 PRINT@32*Y+X,A$:
710 PRINT@64*Y+2*X,A$:
710 PLOT X,Y,A$

```

# Soluzioni dei quesiti

Qui di seguito ci sono alcune possibili soluzioni ai quesiti proposti nel libro. Può darsi che le tue risposte siano diverse, ma ciò non ha importanza, purché funzionino. Controlla però che le tue risposte siano chiare e semplici come quelle che seguono.

## Incubo col Computer (pag. 23)

<pre> ] *  90 LET N\$ = CHR\$(INT(RND*26 + 38))       100 PRINT TAB(5);N\$        220 IF F\$ &lt; &gt; N\$ THEN GOTO 240  ]  230 LET S = S + 10 + (CODE(N\$)—38        190 LET N\$ = CHR\$(INT(RND*26 + 65))       90 LET N\$ = CHR\$(INT(RND(1)*26 + 65))       90 LET N\$ = CHR\$(INT(RND(0)*26 + 65))       1230 LET S = S + 10 + (CODE(N\$)—65)       230 LET S = S + 10 + (ASC(N\$)—65) </pre>	<p>Sceglie una lettera invece di un numero e la visualizza sullo schermo.</p> <p>Confronta il tasto premuto con la lettera scelta.</p> <p>Aumenta il punteggio. (Ciò dipende dalla lettera: Z vale più di A per esempio).</p>
---	---

Nota le diverse versioni per i diversi computer.

## Il Mago dei Numeri (pag. 24)

<pre> 5 LET L = 0 6 LET W = 0  355 LET L = L + 1 360 GOTO 390 380 LET W = W + 1 390 PRINT 400 PRINT "MAGO      TU" 410 PRINT 420 PRINT "    ";L"    ";W 430 FOR Q = 1 TO 1000 440 NEXT Q 450 GOTO 30 </pre>	<p>Azzera il numero di vittorie e di sconfitte all'inizio del gioco.</p> <p>Aumenta il numero di volte che hai perso.</p> <p>Aumenta il numero di volte che hai vinto.</p> <p>Visualizza il tabellone del punteggio.</p> <p>Si ferma in modo che tu possa vedere il punteggio.</p> <p>Torna indietro per iniziare un'altra partita.</p>
---	---

Per i computer più veloci dovrai cambiare questo numero.

## La Donna Ragno (pag. 28)

<pre> 5 LET NG = 0  355 LET NG = NG + 1 360 IF G\$ = T\$ THEN GOTO 370 362 PRINT "NO, SBAGLIATO" 364 LET G = G + 5  366 IF NG = 2 THEN GOTO 410 368 GOTO 230 </pre>	<p>Azzera il numero di tentativi di tirare a indovinare.</p> <p>Aumenta il numero di tentativi di tirare a indovinare.</p> <p>Se hai indovinato, ferma il gioco.</p> <p>Stampa il messaggio che ti comunica che hai sbagliato.</p> <p>Aumenta il numero di tentativi fatti di 5: punizione per non aver indovinato.</p> <p>Se hai usato entrambe le possibilità di tirare a indovinare, il gioco finisce.</p> <p>Torna indietro per il tentativo successivo. (Una pausa ti permette di leggere il messaggio).</p>
---	---





# **GIOCHI DEL MISTERO**

**Jenny Tyler e Chris Oxlade**



## **Indice**

- 40 I programmi dei  
giochi del mistero**
- 41 Torre del Terrore**
- 42 Teschi della Piramide**
- 44 L'Abbraccio del Mostro**
- 46 Lo Squalo (Jaw)**
- 48 Streghe Volanti**
- 50 Micropuzzle**
- 55 Risposte ai quesiti**

## I programmi dei giochi del mistero

I programmi seguenti sono scritti in una versione standard del BASIC, e ci sono le linee di conversione per la maggior parte delle principali marche di home-computer. Cerca sulla sinistra del programma il simbolo del tuo computer e poi vai a cercare la versione adatta di quella linea nella lista delle modifiche. Questi sono i simboli per le varie marche di computer:

- ▲ **Commodore 64 e VIC 20**
- ✱ **BBC ed Electron**
- **Spectrum**
- **Apple**
- **TRS-80**  
(versione di BASIC esteso)

## Qualcosa sui giochi

I giochi sono molto semplici. Il loro scopo è quello di darti una mano affinché tu ti abitui al tuo computer e al linguaggio BASIC, introducendo i listati, correggendoli, e guardando come funzionano. I programmi non contengono né grafica né suono, dato che questi variano molto da computer a computer, ma se vuoi, puoi provare ad aggiungerceli tu.

Puoi fare tanti cambiamenti e adattamenti ai giochi quanti ne vuoi: ci sono suggerimenti sui modi di farlo vicino ad ogni programma, e puoi sperimentare anche con tue idee. In questo modo potrai usare i giochi contenuti in questo libro come una base di partenza per creare i tuoi programmi che saranno più lunghi e complicati.

*Micropuzzle è stato scritto da Les Howarth e l'Abbraccio del Mostro da Adrian Hall.*

*Le illustrazioni sono di Rob McCaig, Sue Walliker, Martin Newton e Graham Smith.*

*Progetto di Roger Priddy*

## Come introdurre i programmi e come farli "girare"

Ricordati che anche programmi corti possono essere molto difficili da copiare correttamente e possono richiedere molto tempo. Controlla ogni linea mentre la scrivi, dato che è molto facile commettere errori anche se si è esperti. Dopo aver introdotto tutto il listato, ricontrollalo per accertarti di non avere saltato linee, spazi o segni di punteggiatura.

Per iniziare il gioco, dai il RUN. Leggi prima la presentazione del gioco, in modo da avere un'idea di ciò che devi fare prima di iniziare. Se il programma non funziona correttamente, è abbastanza probabile che ci sia un errore da qualche parte, perciò dai il LIST e ricontrollalo nuovamente.

Quando la partita è finita, può darsi che il computer ti chieda se vuoi farne un'altra, oppure ti comunichi qualcosa come BREAK IN 200, nel qual caso dovrai dare il RUN per giocare ancora.

## Come cambiare la velocità

Alcuni giochi dipendono sia dalla tua prontezza di riflessi che dalla velocità del tuo computer. Può darsi che tu voglia regolare la velocità in questo caso troverai le istruzioni accanto al listato del programma.

# Torre del Terrore



Hai la bocca secca, le gambe che ti tremano e il cuore che ti batte all'impazzata: sei entrato nella Torre del Terrore ...

Premi il tasto P per spostarti di stanza in stanza. Ah!! c'è uno scheletro, poi un fantasma, poi uno spettro senza testa! Ad ogni nuovo spavento le tue pulsazioni vanno alle stelle. Prosegui (P), oppure ti ritiri (R) per riprendere fiato? Guarda l'ora: hai tempo solo fino a mezzanotte per raggiungere la cima della Torre e il Tesoro agognato. Controlla anche le tue pulsazioni: se raggiungono il valore di 150, diventerai pazzo e niente ti potrà impedire di gettarti dalla finestra.

```
3 REM TORRE DEL TERRORE
5 VERSIONE GENERALE
10 GOSUB 470
20 LET R$="BUONA FORTUNA"
```

Prova a cambiare il limite massimo delle pulsazioni.

```
30 LET RM=0
40 LET H=21:LET M=INT (RND (1)*10)+10
50 LET P=50
60 CLS :PRINT:PRINT
70 PRINT "LA TORRE DEL TERRORE"
80 PRINT "===== "
90 PRINT:PRINT R$
100 LET R$="":LET FL=INT (RM/5)
110 LET R=RM-FL*5+1
120 PRINT:PRINT "TI TROVI"
130 IF FL=0 THEN PRINT "AL PIANTERRENO"
140 IF FL=6 THEN PRINT "AL PIANO PIU' ALTO"
150 IF FL>0 AND FL<6 THEN PRINT "AL "FL"."PIANO"
160 PRINT "NELLA STANZA "R
170 PRINT:PRINT "SONO LE ORE "H":"M":"S"
180 PRINT:PRINT "PULSAZIONI CARDIACHE: "P
190 LET GF=0
200 IF RM=30 THEN GOTO 350
210 LET TR=INT (RND (1)*9+1)
220 IF RND (1)>.6 THEN GOSUB 420
230 PRINT:PRINT "RITORNI INDIETRO O PROSEGUI (R/P)"
240 LET I$=INKEY$
250 IF I$<>"R" AND I$<>"P" THEN GOTO 240
260 IF I$="P" THEN GOSUB 400
270 IF I$="R" THEN LET RM=RM-1:LET P=P-5
280 IF RM=-1 THEN LET AM=0
```

Sei capace di aggiungere stanze?

Prova ad aggiungere altri pericoli e a modificare il valore di shock.

Che cos'è il Tesoro della Torre e come è arrivato lì? Forse puoi aggiungere al programma le tue idee.

```
290 LET M=M+INT (RND (1)*3+1):IF M>59
    THEN LET M=M-60:LET H=H+1
300 IF H=12 THEN GOTO 360
310 IF P>150 THEN GOTO 380
320 IF P<40 THEN LET P=40
330 IF FL=TR AND RND (1)>.5 THEN GOSUB 520
340 GOTO 60
350 PRINT "BENE, IL TESORO E' TUO!":STOP
360 PRINT:PRINT "E' MEZZANOTTE!"
370 PRINT:PRINT "TROPPO TARDI!":STOP
380 PRINT:PRINT "SEI IMPAZZITO E"
390 PRINT "TI SEI GETTATO DALLA FINESTRA!":STOP
400 IF GF=1 THEN LET P=P+S*2:LET R$="AAAAHHH!"
410 LET P=P-1:LET RM=RM+1:RETURN
420 LET TY=INT (RND (1)*3+1)
430 LET W$=G$(TY)
440 LET S=INT (RND (1)*5)+FL+TY*2
450 PRINT "D'AVANTI A TE VEDI":PRINT W$
460 LET GF=1:RETURN
470 DIM G$(3)
480 LET G$(1)="UNO SCHELETRO"
490 LET G$(2)="UN FANTASMA"
500 LET G$(3)="UNO SPETTRO SENZA TESTA"
510 RETURN
520 LET R$="SEI CADUTO IN UNA BOTOLA!":LET RM=RM-5:LET P=P+10:RETURN
```

## Come funziona

10: Va alla sezione di inizializzazione per leggere i dati e immagazzinarli.

30: RM è la stanza in cui ti trovi.

40-50: Stabilisce l'ora e il numero iniziale di pulsazioni.

90: Visualizza i messaggi immagazzinati in RS.

100-160: Calcola il piano e la stanza in cui ti trovi, tramite il valore di RM, e visualizza queste informazioni.

170-180: Visualizza altre informazioni.

190: GF è una variabile "flag" che viene attivata quando hai davanti qualche pericolo.

200: Controlla se hai raggiunto l'ultima stanza.

210: Sceglie la stanza dove mettere la botola.

220: Possibilità casuale che ci sia qualcosa davanti a te: se c'è, va alla subroutine appropriata.

230-280: Aspetta la risposta del giocatore e agisce di conseguenza.

290: Tiene conto del trascorrere del tempo.

300: Controlla se è mezzanotte.

310: Controlla se le pulsazioni sono troppo alte.

330: Controlla se ti trovi nella stanza con la botola.

400-410: Routine di proseguimento. Aumenta le pulsazioni se c'è un pericolo.

420-460: Sceglie il tipo di pericolo e il suo valore di shock.

470-510: Lettura dei dati per iniziare il gioco (inizializzazione).

520: Subroutine della botola.


## Linee di conversione

- 40,210,220,290,330,420,440 Sostituisci RND(1) con RND
- 40,210,220,290,330,420,440 Sostituisci RND(1) con RND(0)
- ▲ 60 PRINT CHR\$(147):PRINT:PRINT
- 60 HOME:PRINT:PRINT
- 240 LET I\$=INKEY\$(0)
- ▲ 240 GET I\$
- 240 I\$="" IF PEEK(-16384)>127 THEN GET I\$
- 470 DIM G\$(3,15)

# Teschi della Piramide

Una mente senza corpo: ecco cosa sei diventato dopo quel terribile incidente di molti anni fa, durante la costruzione della Piramide.

Per soddisfare le richieste del Capo degli Spiriti Malvagi ci vuole tutta la tua energia mentale. Egli ti forza un numero in testa ogni giorno; devi far cadere con la tua volontà dei teschi dalla Piramide finché non raggiungono le pietre numerate ai piedi della stessa, nel tentativo di raggiungere quel numero. Per far cadere un cranio, premi 1, 2, 3 o 4. Devi usare cinque crani per volta, né più né meno. Raggiungi il numero giusto e liberati dalla Maledizione della Piramide.



Aumenta il valore di SP alla linea 10 per rendere il programma più veloce.

```
3 REM I CRANI DELLA PIRAMIDE
5 REM VERSIONE GENERALE
10 LET TS=0:LET I$="0":LET SP=3
20 LET N=INT (RND (1)*30)+10
30 CLS:PRINT
40 PRINT "I CRANI DELLA PIRAMIDE"
50 PRINT
60 PRINT "IL NUMERO FINALE DA"
70 PRINT "OTTENERE E' :";N:PRINT
80 LET K=3000:GOSUB 540
90 CLS:LET Y=2
100 FOR I=1 TO 4
110 LET A$=STR$ (I):LET X=6+2*I
120 GOSUB 560
130 NEXT I
140 LET A$="↑":FOR I=1 TO 5
150 LET Y=2+2*I
160 FOR J=1 TO 1+3
170 LET X=7-1+2*J
180 GOSUB 560
190 NEXT J
200 NEXT I
210 LET Y=14
220 FOR I=1 TO 9:LET A$=STR$ (I)
230 LET X=1+2*I:GOSUB 560
240 NEXT I
250 LET S=0
260 FOR I=1 TO 5
270 LET A$="SCEGLI QUALE FILA"
280 LET X=0:LET Y=0:GOSUB 560
290 LET I$=INKEY$
```


## Come funziona il programma

10: Inizializza le variabili.

20: Sceglie il numero che il giocatore dovrà raggiungere.

30-80: Visualizza il messaggio iniziale e aspetta che il giocatore lo veda

90-240: Visualizza la piramide e i numeri alla base e alla sommità di essa.



Prova a sostituire il simbolo del cranio con un carattere grafico.

Se ci riesci, aggiungi dei rumori terribili che accompagnino la caduta dei teschi dalla Piramide.

250: Azzerà il punteggio.

260: Prepara il numero di crani.

270-280: Visualizza il messaggio in alto a sinistra dello schermo.

290-300: Aspetta che venga premuto un tasto.

310-320: Controlla che il tasto premuto sia un numero compreso tra 1 e 4, se non lo è, torna indietro e ne aspetta un altro.

330: X e Y determinano la posizione del cranio. Questa linea li pone in cima alla piramide.

340: F è una variabile 'flag' che registra se la mossa precedente è stata un rimbalzo laterale.



Scommetto che non sei capace di calcolare come si fa a dare un numero maggiore di teschi per ogni giro.



```

300 IF I$="" THEN GOTO 290
310 LET R=VAL (I$)
320 IF R<1 OR R>4 THEN GOTO 290
330 LET Y=3:LET K=400:LET X=6+2*R
340 LET F=0
350 LET A$=""
360 GOSUB 560
370 GOSUB 540
380 LET A$="" :GOSUB 560
■□390 IF F=0 AND Y<13 THEN LET X=X+SGN (RND (1)-.5)
400 LET Y=Y+1:LET F=1-F
410 IF Y<14 THEN GOTO 350
420 LET S=S+(X-1)/2
430 LET X=15:LET Y=0
440 LET A$=STR$ (I)+": "+STR$ (S)
450 GOSUB 560
460 NEXT I
470 LET K=1000:GOSUB 540
480 LET J=ABS (S-N):LET TS=TS+D
●▲490 CLS:PRINT:PRINT
500 PRINT "LA DIFFERENZA E' ":TS
510 IF D>1 THEN LET K=3000:GOSUB 540:GOTO 20
520 PRINT "SEI LIBERO. PUOI ANDARTENE"
530 STOP
540 FOR T=1 TO K STEP SP:NEXT T
550 RETURN
□■●▲560 PRINT TAB(X,Y);A$
570 RETURN

```

**350-370:** Visualizza il cranio e aspetta.

**380:** Cancella il cranio.

**390:** Sceglie da che parte effettuare il rimbalzo se il valore di F permette un rimbalzo laterale.

**400:** Sposta il cranio di una riga verso il basso.

**410:** Controlla se il cranio ha toccato terra.

**420:** Calcola il punteggio a seconda della colonna in cui è finito il cranio.

**430-450:** Visualizza il punteggio.

**470:** Aspetta.

**480-500:** Somma la differenza tra il punteggio e il numero richiesto al punteggio totale.

**510:** Se la differenza è maggiore di 1, torna indietro per un altro gioco.

**520:** Visualizza il messaggio di vittoria.

**540-550:** Subroutine per far fare al computer una pausa di lunghezza dipendente da K.


**560-570:** Subroutine per spostare il cursore alla posizione X, Y dello schermo.



Non credere di essere abbastanza intelligente da riuscire a cambiare i limiti del numero da raggiungere.

## Linee di conversione

- 20,390 Sostituisci RND(1) con RND
- 20,390 Sostituisci RND(1) con RND(0)
- 30,90,490 Sostituisci CLS con HOME
- ▲ 30,90,490 Sostituisci CLS con PRINT CHR\$(147)
- ▣ 290 LET I\$=INKEY\$(0)
- 290 I\$="":IF PEEK(-16384)>127 THEN GET I\$
- ▲ 290 GET I\$
- 560 PRINT AT Y,X;A\$
- 560 VTAB(Y+1):HTAB(X+1):PRINT A\$
- ▲ 560 PRINT CHR\$(19)::FOR LL=1 TO Y:PRINT:NEXT LL
- 560 PRINT@Y\*32+X,A\$



# L'abbraccio del Mostro

Lottare coi mostri è uno sport per lunatici! Il che non vuol dire molto per te, che sei la mente cui è affidato il compito di tenere a bada questo mucchio di ossa e muscoli, che è uno dei più schifosi mostri dell'Universo.

Nella tua qualità di mente, devi fare molti calcoli veloci e precisi. Devi per esempio calcolare lo sforzo muscolare necessario per sfuggire al mostro, e questo implica il dover moltiplicare le dimensioni del mostro per la distanza a cui esso si trova da te. Se i numeri ti sembrano troppo complicati, premi il Bottone del Panico (tasto P). Dovrai poi calcolare quanto adrenalina occorre al corpo per superare una crisi, dividendo il battito cardiaco richiesto per l'ossigeno disponibile. Stai attento tuttavia, in quanto un abuso del Bottone del panico sforza troppo il cuore e potrebbe causarti un crollo alla fine. Per poter fare un'altra partita, devi sopravvivere a 12 round nella lotta contro il mostro.

## Come funziona

20: Azzera il numero di volte che è stato usato il tasto P.

30-60: Fissa i valori massimi della forza del mostro e della distanza.

70: Determina il numero di round.

80: Dimensiona il mostro.


90: Determina la distanza del mostro.

```
10 REM LOTTA COL MOSTRO
20 LET P=0
30 LET K=3
●▲40 CLS
50 LET X=1
60 LET Y=6
70 LET N=-1
75 LET N=N+1
■□80 LET G=INT (RND (1)*Y+X)
■□90 LET I=INT (RND (1)*K+K)
100 LET Y=Y+0.5
110 LET X=X+0.5
120 LET K=K+0.5
130 PRINT
140 PRINT
150 PRINT "DIMENSIONI DEL MOSTRO: ";
160 PRINT G
170 PRINT
180 PRINT "DISTANZA DAL MOSTRO : ";
190 PRINT I
200 PRINT
210 PRINT "SFORZO MUSCOLARE ?      ";
220 GOSUB 570
230 IF Z<>G*I THEN GOTO 320
●▲240 CLS
250 PRINT "HAI TENUTO TESTA AL MOSTRO"
260 IF N<11 THEN GOTO 75
270 PRINT "FIUU!!! - IL MOSTRO "
280 PRINT "E' STANCO ED E' ANDATO A"
290 PRINT "CERCARSI UN'ALTRA VITTIMA."
300 PRINT "SEI VIVO PER MIRACOLO!"
310 STOP
●▲320 CLS
330 PRINT "LE POSSENTI BRACCIA"
340 PRINT "DEL MOSTRO HANNO FATTO "
350 PRINT "POLPETTE DI TE."
360 PRINT
370 PRINT "SEI SOPRAVVISSUTO ";N;" ROUND"
380 STOP
●▲390 CLS
■□400 LET WX=INT (RND (1)*9+1)
■□410 LET WY=INT (RND (1)*9+1)
420 LET W=WX*WY
430 LET P=P+1
440 IF P=4 THEN GOTO 700
450 IF P=3 THEN PRINT "STAI VEDENDO LE STELLE!"
460 PRINT "SEI IN PREDI AL PANICO!"
470 PRINT
480 PRINT "BATTITO CARDIACO: ";W
490 PRINT "OSSIGENO DISPONIBILE: ";WX
500 PRINT
510 PRINT "QUANTITA' DI ADRENALINA? ";
■□520 LET Q=100
530 GOSUB 580
540 IF Z<>WY THEN GOTO 320
550 CLS:IF N<11 THEN GOTO 75
560 GOTO 270
570 LET Q=0
■□580 LET Z#=""
```

Se continui a perdere, ti converrà rallentare il programma aumentando il valore di Q alla linea 660.

Diminuisci Q, se il gioco ti sembra troppo facile.





```

■●▲590 LET A$=INKEY$
□ 600 IF A$=CHR$(13) THEN GOTO 680
□ 610 IF A$="P" THEN GOTO 390
620 IF VAL(A$)=0 AND A$<"0" THEN GOTO 650
630 PRINT A$
640 LET Z$=Z$+A$
650 LET Q=Q+1

■● 660 IF Q=500 THEN GOTO 320
670 GOTO 590
680 LET Z=VAL(Z$)
690 RETURN
▲700 CLS
710 PRINT "SEI CROLLATO"
720 STOP

```

Per rendere il gioco più difficile, prova a indovinare come si possa aumentare il numero di round.

## Come funziona il programma (seguito)

100-120: Aumenta di 0,5 i limiti delle dimensioni e della distanza, ad ogni round.

210-220: Aspetta una risposta e va alla subroutine per controllarla.

230: Se la risposta è sbagliata, salta al messaggio di sconfitta.

250: Se hai vinto, visualizza il messaggio appropriato.

260: Torna per il round successivo.

270-310: Messaggio di vittoria se si superano i 12 round.

330-380: Messaggio di sconfitta.

400-410: Decide la quantità di ossigeno e di adrenalina necessaria.

420: Determina il valore dei battiti cardiaci.

430-450: Conta quante volte sei ricorso al Botone del Panico; avverte dopo la terza volta e termina il gioco dopo la quarta.

460-510: Stampa i messaggi di Panico e aspetta la risposta.

520: Pone Q al valore di inizio del conteggio, che viene effettuato dalle linee 650 e 660.

530: Va alla subroutine di risposta.

540: Se la risposta è sbagliata, torna al messaggio di sconfitta.

550-560: Se la risposta è esatta, va al round successivo o al messaggio di vittoria.

570-690: SUBROUTINE PER LA RISPOSTA.

570: Azzera il conteggio.

580: Riserva spazio in memoria per Z\$.

590: Controlla se è stato premuto un tasto.

600: Controlla se il tasto premuto è RETURN (CHR\$(13)), nel qual caso va alla linea 680.

610: Controlla se il tasto premuto è P; se sì, va alla routine "Panico".

620: Se non è stato premuto un tasto, aumenta il valore di Q.

630: Se il tasto premuto non è né RETURN né P, lo visualizza.

640: Z\$ contiene i tasti premuti nel corso del round.

650-660: Loop per il conteggio del tempo. Se Q raggiunge il valore di 1000, senza che sia stato premuto un tasto, va al messaggio di sconfitta.

670: Torna indietro per la successiva pressione di un tasto.

680: Converti i tasti numerici in Z\$ in numeri, per metterli in Z.

710: Messaggio di sconfitta per abuso del tasto P.

## Linee di conversione

- ▲ 40,240,320,390,550,700 Sostituisci CLS con PRINT CHR\$(147)
- 40,240,320,390,550,700 Sostituisci CLS con HOME
- 80,90,400,410 Sostituisci RND(1) con RND
- 80,90,400,410 Sostituisci RND(1) con RND(0)
- ✱ 520 LET Q=400
- 580 LET Z\$=""
- ✱ 590 LET A\$=INKEY\$(1)
- ▲ 590 GET A\$
- 590 A\$="" : IF PEEK(—16384)>127 THEN GET A\$
- 595 IF A\$="" THEN GOTO 650
- 597 PAUSE 40
- 600 IF A\$CHR\$(13) THEN PAUSE 20:GOTO 680
- 610 IF A\$="P" THEN PAUSE 30:GOTO 390
- 630 PRINT A\$:PAUSE 15
- ✱ 660 IF Q=1000 THEN GOTO 320
- 660 IF Q=250 THEN GOTO 320

# Lo Squalo

Mmm ... come sono squisiti gli esseri umani! Ma se tu hai un Cacciatore perennemente alle calcagna, diventa difficile riuscire ad avvicinarli abbastanza da poterli catturare. Tu sei J (Jaw) e ti puoi muovere usando i tasti A, Z, M ed N. Quante P (persone) riesci a mangiare, prima di essere catturato da C (Cacciatore)? C'è un trabocchetto, ovviamente: ogni volta che riesci a catturare e divorare un delizioso essere umano, ti emozioni tanto da dimenticarti a cosa corrispondono i tasti.

## Come funziona

10: Riserva spazio in memoria per il reticolato; va alla subroutine per leggere i dati relativi ad esso.

20: Azzerà le variabili per la partenza.

30: Fissa ogni quanto si muoverà il cacciatore.

40-50: Stabilisce la direzione del movimento relativo ai vari tasti.

60: Pone Lo Squalo (Jaws) alla posizione iniziale.

70: Va alla subroutine che visualizza il reticolo.

80: Va alla subroutine che fissa le posizioni del cacciatore e delle persone.

90: Aggiorna la posizione dello squalo.

100-140: Calcola la nuova posizione dello squalo a seconda del tasto premuto.

150: Controlla che questa nuova posizione non sia un muro.

160: Se lo è, non sposta J.

170-180: Stampa uno spazio vuoto nella posizione precedente.

190-200: Stampa una J per indicare la nuova posizione dello squalo.

210: Sposta lo squalo nella nuova posizione.

220: Controlla se J ha attaccato qualche P e salta, se del caso, alla subroutine.

230: Controlla se J ha attaccato il Cacciatore, e se sono nella stessa posizione fa finire il gioco.

240: Muove il cacciatore.

250-260: Visualizza il punteggio sinora ottenuto.

270: Aumenta il numero di mosse e torna indietro alla mossa successiva.

280: Fa una pausa in modo che tu possa vedere se sei stato catturato.

```


10 DIM M$(12):DIM K$(4):GOSUB 660
20 LET S=0:LET T=0:LET G=0
30 LET U=1
40 LET K$(1)="A":LET K$(2)="M"
50 LET K$(3)="Z":LET K$(4)="N"
60 LET PX=2:LET PY=2
70 GOSUB 600
80 GOSUB 490:GOSUB 440
90 LET NX=PX:LET NY=PY
100 LET I$=INKEY$
110 IF I$=K$(1) THEN LET NY=NY-1
120 IF I$=K$(2) THEN LET NX=NX+1
130 IF I$=K$(3) THEN LET NY=NY+1
140 IF I$=K$(4) THEN LET NX=NX-1
150 LET X=NX:LET Y=NY:GOSUB 530
160 IF F=1 THEN GOTO 230
170 LET X=PX:LET Y=PY:LET A$=" "
180 GOSUB 640
190 LET X=NX:LET Y=NY:LET H$="J"
200 GOSUB 640
210 LET PX=NX:LET PY=NY
220 IF PX=TX AND PY=TY THEN GOSUB 340
230 IF PX=GX AND PY=GY THEN GOSUB 280
240 IF RND(1)>0 THEN GOSUB 370
250 LET X=14:LET Y=12:LET A$=STR$(T)
260 GOSUB 640
270 LET G=G+1:GOTO 90
280 FOR P=1 TO 2000:NEXT P
    
```

## Linee di conversione

```

10 DIM M$(12,16):DIM K$(4):GOSUB 660
100 LET I$=INKEY$(0)
100 I$="":IF PEEK(-16384)>127 THEN GET I$
100 GET I$
    
```

Ti sfido a far muovere più spesso il cacciatore.



```

240,390,400,550,560 Sostituisci RND(1) con RND
240,390,400,550,560 Sostituisci RND(1) con RND(0)
290,600 Sostituisci CLS con HOME
290,600 Sostituisci CLS con PRINT CHR$(147)
530 LET F=0:IF M$(Y,X)<>" " THEN LET F=1
640 PRINT AT Y,X:A$
640 VTAB(Y):HTAB(X):PRINT A$
640 PRINT CHR$(19):FOT LL=1 TO Y:PRINT:PRINT
TAB(X):A$
640 PRINT@ Y*32+X,A$:
    
```







```

●▲290 CLS:PRINT "SEI STATO PRESO!"
300 LET S=INT (10000*(T/G)):PRINT "PUNTEGGIO "S
310 PRINT:PRINT "UN'ALTRA PARTITA? (S/N)"
320 INPUT A$:IF A$="S" THEN GOTO 20
330 STOP

```

290-330: Stampa un messaggio e il punteggio, se Jaws è stato preso. Chiede se il giocatore vuole fare un'altra partita.



Il punteggio dipende dal numero di persone e dal numero di mosse.

```

340 LET T=T+1:LET U=U-0.03
350 GOSUB 390:GOSUB 490
360 RETURN
370 LET X=GX:LET Y=GY:LET A$=" "
380 GOSUB 640:GOSUB 440:RETURN
■□390 LET K1=INT (RND (1)*4)+1
■□400 LET K2=INT (RND (1)*4)+1
410 LET T$=K$(K1)
420 LET K$(K1)=K$(K2):LET K$(K2)=T$
430 RETURN
440 GOSUB 550
450 IF X=TX AND Y=TY THEN GOTO 440
460 LET GX=X:LET GY=Y
470 LET A$="C":GOSUB 640
480 RETURN
490 GOSUB 550
500 LET TX=X:LET TY=Y
510 LET A$="P":GOSUB 640
520 RETURN
□530 LET F=0:IF MID$(M$(Y),X,1)>" " THEN LET F=1
540 RETURN
■□550 LET X=INT (RND (1)*14)+2
■□560 LET Y=INT (RND (1)*9)+2
570 GOSUB 530
580 IF F=1 THEN GOTO 550
590 RETURN
●▲600 CLS:LET X=1
610 FOR Y=1 TO 12:LET A$=M$(Y)
620 GOSUB 640:NEXT Y
630 RETURN
■●▲640 PRINT TAB(X,Y):A$
□
650 RETURN
660 FOR I=1 TO 12:READ M$(I):NEXT I
670 RETURN
680 DATA "0000000000000000"
690 DATA "0 0 0 0 0 0 0 0 0 0 0 0"
700 DATA "0 0 000 00 000 0"
710 DATA "0 0 0 0 0 0 0 0 0 0 0 0"
720 DATA "0 0 0 0000 0 0 0"
730 DATA "000 0 0 0 0 0 0"
740 DATA "0 0 0 0 0 0 0 0 0 0 0 0"
750 DATA "0 0 00 0 0 0 0 0 0 0 0 0"
760 DATA "0 0 00 0 000 0 0 0"
770 DATA "0 0 0 0 0 0 0 0 0 0 0 0"
780 DATA "0000000000000000"
790 DATA "MANGIATI:"

```

340-360: Subroutine che aumenta il punteggio e le possibilità di mosse del cacciatore; va ad un'altra subroutine per cambiare le direzioni cui si riferiscono i tasti per lo spostamento; posiziona la persona successiva.

370-380: Subroutine che toglie il cacciatore dalla posizione precedente per visualizzarlo in quella successiva.

390-430: Subroutine che cambia i tasti dei movimenti.

440-480: Subroutine che visualizza il cacciatore, utilizzando un'altra subroutine per scegliere a caso la posizione in cui metterlo. La linea 450 controlla che nella posizione scelta non ci sia già una persona.

490-520: Subroutine per la visualizzazione di una persona, utilizzando un'altra subroutine per la scelta casuale della posizione dove porla.


530-540: Subroutine che controlla se la posizione X, Y coincide con un muro. Se sì, pone F uguale ad 1.

550-590: Subroutine che fissa una posizione a caso e controlla che non sia un muro.

600-630: Subroutine per la visualizzazione del reticolo.

640-650: Subroutine per visualizzare in posizione X, Y.

660-790: Subroutine per la lettura dei dati relativi al reticolo.



Queste sono le linee per creare il reticolo. Potresti modificarle, disegnando un nuovo reticolo su un foglio a quadretti e copiandolo nel programma. Devi usare lettere O per i muri e spazi per i passaggi.



# Streghe volanti



Sei l'assistente-capo di Superstrega, il tuo compito è quello di volare in giro per la terra, raccogliendo ingredienti freschi per le sue pozioni puzzolenti. Saprai esattamente che cosa raccogliere in quanto essa trasmette direttamente sullo schermo del tuo computer le sue ributtanti ricette. Vedrai apparire sotto di te mentre voli ingredienti come D (Dita di Tritone) e P (Pipistrelli). Devi premere un tasto per piombare su uno di essi e raccogliarlo (per farlo, dovrai atterrare proprio sopra). Fai attenzione, che qualcuno degli oggetti volanti potrebbe essere il Distruggi-Manici-di-Scopa travestito: se atterri su di esso, perderai il tuo manico di scopa. (Si inizia il gioco con tre manici di ricambio). Devi raccogliere tutto ciò che ti si richiede, prima di arrivare al Calderone, altrimenti Superstrega non sarà soddisfatta ...

```
3 REM STREGHE VOLANTI
5 REM VERSIONE GENERALE
10 GOSUB 700:GOSUB 620
20 LET NW=4:LET M=0:LET FM=50
30 GOSUB 530:LET H=NH
40 GOSUB 550
50 LET EF=0:LET FF=0:LET HH=H
60 FOR I=1 TO 4:LET G(I)=0:NEXT I
70 FOR I=1 TO 4:LET A$=P$(I)+":":LET X=1
80 LET Y=1+I:GOSUB 610:NEXT I
90 GOSUB 490:GOSUB 430
100 IF M=FM THEN GOTO 250
110 IF EF=1 THEN GOTO 230
120 IF INKEY$(0)=" " THEN GOTO 90
130 LET HH=H-1:GOSUB 510
140 GOSUB 490:GOSUB 430
150 LET H=HH
160 IF H>0 THEN GOTO 130
170 IF P=17 THEN GOSUB 320
180 IF FF=1 THEN GOTO 240
190 GOSUB 530
```




È troppo difficile per te? Fallo diventare più facile aumentando il valore di FM alla linea 20. Così metti più oggetti tra la partenza e il calderone.

```
200 LET HH=H+1:GOSUB 510:GOSUB 490
210 LET H=HH:IF H=NH THEN GOTO 90
220 GOSUB 430:GOTO 200
230 LET A$="SUPERSTREGA SI CONGRATULA CON TE"
240 LET A$="I MANICI DI SCOPA SONO FINITI":GOTO 260
```

```
250 LET A$="TROPPO TARDII"
260 LET X=1:LET Y=15:GOSUB 610
270 STOP
280 LET A$="W":LET X=P:GOSUB 610
290 FOR T=1 TO 100:NEXT T
300 LET NW=NW-1:IF NW=0 THEN LET FF=1
310 RETURN
320 IF W=1 THEN GOSUB 280:RETURN
330 LET G(R)=G(R)+1
340 LET A$=STR$(G(R)):LET X=17
350 LET Y=1+R:GOSUB 610
360 LET P=19:GOSUB 430
370 LET F=0
380 FOR I=1 TO 4
```

```
390 IF G(I)=N(I) THEN LET F=F+1
400 NEXT I
```



Prova a sostituire le lettere con simboli grafici.

Perché non cambiare gli ingredienti?

## Come funziona

20: Fissa i valori delle variabili. (Prova a capire a che cosa si riferiscono).

30: H è l'altezza da terra a cui stai volando. NH è la tua nuova altezza dopo una picchiata.

50: EF e FF sono variabili 'flag' che vengono attivate quando il gioco è finito. HH è una variabile per ottenere altezze intermedie mentre sali o scendi.

60: G0 tiene conto di quanti di ciascun tipo di oggetti hai raccolto.

70-80: Visualizza la ricetta.

90-120: Sposta la linea di terra e gli oggetti sullo schermo finché non viene premuto un tasto.

130-160: Continua a muovere il suolo e gli oggetti e muove la strega verso il basso fino al livello del suono.

170: Se P=17, la strega è atterrata su un oggetto.

180: Controlla la variabile 'flag' della fine di gioco.

200-220: Muove il suolo e gli oggetti finché la strega non raggiunge la nuova altezza.

230-270: Messaggi di fine partita.

280-310: Subroutine che mostra una W se l'oggetto è il Distruggi-Manici-di-Scopa travestito. Aumenta NW e attiva FF se tutti i manici di scopa sono stati utilizzati.

320: Subroutine che controlla se sei atterrato sul Distruggi-Manici-di-Scopa.

330-350: Aumenta il numero di oggetti R raccolti e stampa sullo schermo il nuovo totale.

370-420: Controlla se tutti gli ingredienti per questa ricetta sono stati raccolti.



```

410 IF F=4 THEN LET EF=1
420 RETURN
430 LET GD=GD+1:IF GD=6 THEN LET GD=1
440 LET A$=G$(GD):LET X=1:LET Y=15:GOSUB 610
450 LET A$=" ":LET X=P:LET Y=14:GOSUB 610
460 LET P=P+1:IF P=20 THEN GOSUB 550
470 LET X=P:LET A$=C$:GOSUB 610
480 RETURN
490 LET A$="-Y-":LET X=16:LET Y=14-HH
500 GOSUB 610:RETURN
510 LET A$=" ":LET X=16:LET Y=14-H
520 GOSUB 610:RETURN
530 LET NH=INT (RND (1)*5+3)
540 RETURN
550 LET R=INT (RND (1)*4)+1
560 LET W=0:IF RND (1)>0.8 THEN LET W=1
570 LET C$=LEFT$ (P$(R),1):LET P=1
580 LET M=M+1
590 IF M=FM THEN LET C$="CALDERONE"
600 RETURN
610 PRINT TAB(X,Y):A$:RETURN
620 CLS:PRINT:PRINT
630 PRINT "LA POZIONE DEVE AVERE":PRINT
640 FOR I=1 TO 4
650 LET N=INT (RND (1)*3)+2:LET N(I)=N
660 PRINT:PRINT :N: " ":P$(I)
670 NEXT I:PRINT:PRINT
680 PRINT "PREMI RETURN PER CONTINUARE"
690 INPUT X$:CLS:RETURN

```

**430-480:** Visualizza la porzione successiva di terra e muove l'oggetto alla posizione P. Se l'oggetto ha superato la strega (cioè P=20), sceglie un altro oggetto.

**490-500:** Visualizza la strega.

**510-520:** Cancella la strega con una serie di spazi.

**530-540:** Determina la nuova altezza per la strega.

**550:** Sceglie un oggetto.

**560:** Possibilità casuale che l'oggetto possa essere il Distruggi-Manici-di-Scopa travestito.

**570:** Mette la prima lettera dell'oggetto in C\$ e pone uguale a 1 la posizione P.

**580:** Aumenta il valore del contatore d'oggetti M.

**590:** Se M equivale al numero massimo permesso di oggetti, C\$ viene cambiata in 'CALDERONE'.

**610:** Subroutine generica per muovere il cursore in posizione pronta per la stampa.

**620-690:** Sceglie la ricetta e la visualizza.

**700-770:** Prepara i dati per iniziare il programma. P\$ contiene i nomi degli oggetti; H\$ contiene i simboli per il suolo. La linea 750 sposta di uno spazio alla volta i simboli.

```

700 DIM P$(4):DIM N(4):DIM G(4)
710 DIM G$(5)
720 FOR I=1 TO 4 READ P$(I):NEXT I
730 DATA "DITA DI TRITONE","OCCHI DI LUPO","RADICI DI FIERA"
735 DATA "PIPISTELLI"
740 LET H$="####"
750 FOR I=1 TO 5:LET G$(6-I)=MID$(H$,I,1)+20:
760 NEXT I:LET GD=1
770 RETURN

```

Che cosa ti farà Superstrega se non ce la fai? Prova ad aggiungere al gioco commenti e penalità.

### Linee di conversione

```


■ 120 IF INKEY$="" THEN GOTO 90
▲ 120 GET I$:IF I$="" THEN GOTO 90
● 120 I$="":IF PEEK(-16384)>127 THEN GET I$
● 125 IF I$="" THEN GOTO 90
★ 125 *FX15,1
530,550,560,650 Sostituisci RND(1) con RND
■ 530,550,560,650 Sostituisci RND(1) con RND(0)
570 LET C$=P$(R,1):LET P=1
610 PRINT AT Y,X:A$:RETURN

```

```

▲ 610 PRINT CHR$(19):FOR LL=1 TO Y:PRINT:
NEXT:PRINT TAB(X):A$:RETURN
● 610 VTAB(Y):HTAB(X):PRINT A$:RETURN
■ 610 PRINT @ Y*32+X,A$:RETURN
▲ 620,690 Sostituisci CLS con PRINT CHR$(147)
● 620,690 Sostituisci CLS con HOME
700 DIM P$(4,15):DIM N(4):DIM G(4)
710 DIM G$(5,24)
750 FOR I=1 TO 5:LET G$(6-I)=H$(I TO I+20)

```



## Micropuzzle

Che cos'è successo? Dove ti trovi? Tutto sembrerebbe normale, se non fosse per il fatto che ti senti un po' strano. Sarà meglio se ti guardi intorno e vedi se riesci a scoprire che cosa sta succedendo. (Suggerimento: prova a parlare al tuo computer usando frasi composte da due parole).

Questa è una specie di mini 'adventure'. Se non hai mai giocato ad un'avventura prima d'ora, sappi che lo scopo del gioco, è quello di scappare.



### Come funziona

Linee 30-360: SEZIONE PRINCIPALE DEL GIOCO.

10 GOSUB 1570

10: Va alla subroutine e mette in memoria i dati.

20 RESTORE:FOR I=1 TO R:READ D\$:NEXT I  
 ●▲ 30 CLS:PRINT "MICRO PUZZLE"  
 40 PRINT "=====  
 50 PRINT "TI TROVI ";D\$

20-50: Prende la descrizione di una stanza e la visualizza.

60 IF R=20 AND F(9)=0 THEN PRINT "SEI AFFRONTATO DA UN GROSSO GATTO"

70 FOR I=1 TO G  
 80 IF L(I)=R AND F(I)=0 THEN PRINT "QUI C'È ";O\$(I)  
 90 NEXT I

70-90: Controlla se c'è qualcosa.

100 PRINT:PRINT "PUOI ANDARE A "  
 110 FOR I=1 TO LEN (R\$(R))  
 120 PRINT MID\$(R\$(R),I,1);";"  
 130 NEXT I

100-130: Trova le direzioni in cui ti puoi muovere e le visualizza.

140 PRINT:PRINT "-----"

150 PRINT M\$:LET M\$="CUME?"

150: Messaggio di errore.

160 IF F(16)=1 THEN PRINT "AUTODISTRUZIONE - CONTO ALLA ROVESCIA: ";L

170-220: Prende le tue istruzioni e le divide in due parole.

170 PRINT "COSA VUOI FARE ORA ";INPUT Q\$  
 180 LET V\$="" :LET W\$="" :LET VB=0 :LET OB=0 :LET LI=LEN (Q\$)  
 190 FOR I=1 TO LI  
 200 IF MID\$(Q\$,I,1)=" " AND V\$="" THEN LET V\$=LEFT\$(Q\$,I-1)  
 210 IF MID\$(Q\$,I+1,1)<>" " AND V\$<>" " THEN LET W\$=RIGHT\$(Q\$,LI-I):LET I=LI  
 220 NEXT I

```

230 IF W$="" THEN LET V$=Q$
240 FOR I=1 TO V
250 IF V$=H$(I) THEN LET VB=I
260 NEXT I

```

**230-260: Controlla che la prima parola sia un verbo.**

```
270 GOSUB 440
```

**270: Va alla subroutine per il controllo della seconda parola.**

```

280 IF VB=0 THEN LET VB=V+1
290 IF W$>"" AND OB=0 THEN LET M$="SAREBBE STUPIDO"
300 IF W$="" THEN LET M$="MI SERVONO DUE PAROLE"
310 IF VB>V AND OB>0 THEN LET M$="NON PUOI "+Q$
320 IF VB>V AND OB=0 THEN LET M$="NON HA SENSO"
330 IF VB<V AND OB>0 AND OB<=G THEN LET M$="NON HAI "+W$

```

**280-330: Prepara messaggi se necessario.**

```
340 LET L=L-1
```

**340: Contatore.**

```

350 IF VB>14 THEN GOTO 380
360 ON VB GOSUB 480,550,550,550,550,550,550,760,790,790,860,920,970
370 GOTO 390
380 ON VB-14 GOSUB 1080,1100,1180,1250,1320,1340,1370,
1400,1460,1510,1560

```

**350-380: Va ad una data subroutine a seconda del verbo che hai introdotto.**

```

390 IF F(14)=0 AND L>0 THEN GOTO 20
400 IF L<1 THEN LET M$="IL TEMPO E' SCADUTO. L'INGRANDITORE SI E' AUTODISTRUTTO!"
410 PRINT M$:PRINT "FINE DELLA PARTITA"
420 IF L>0 THEN GOSUB 1510
430 STOP

```

**390-430: Fine del gioco.**

Scrivi 'INVENTARIO' se vuoi sapere cos'hai con te.

```

440 FOR I=1 TO W
450 IF W$=O$(I) THEN LET OB=I
460 NEXT I
470 RETURN

```

**440-470: Subroutine che controlla se la seconda parola è presente nell'elenco degli oggetti del computer.**

```

480 PRINT "HAI CON TE: "
490 FOR I=1 TO G
500 IF L(1)=25 THEN PRINT O$(I);",";
510 NEXT I
520 LET M$="":PRINT
530 GOSUB 1550
540 RETURN

```

**480-540: Subroutine che ti dice che cos'hai con te.**



```

550 LET D=0
560 IF OB=0 THEN LET D=VB-2
570 IF OB>24 THEN LET D=OB-24
580 IF R=19 AND D=6 THEN LET D=4
590 IF R=20 AND D=6 THEN LET D=4
600 IF R=20 AND D=5 THEN LET D=3
610 IF F(9)=0 AND R=20 AND D=4 THEN LET M$="IL GATTO NON TI LASCIA":RETURN
620 IF R=2 AND L(1)=25 AND D=3 THEN LET M$=M1$:RETURN
630 IF R=7 AND F(3)=0 THEN LET M$="UN TOPO FURIOSO TI SBARRA IL PASSO":RETURN
640 LET F(13)=0:LET RL=LEN (R$(R))
650 FOR I=1 TO RL
660 LET U$=MID$(R$(R),I,1)
670 IF (U$="N" AND D=1 AND F(13)=0) THEN LET R=R-6:LET F(13)=1
680 IF (U$="S" AND D=2 AND F(13)=0) THEN LET R=R+5:LET F(13)=1

```

**550-570: Subroutine che esegue le tue istruzioni in merito alla direzione nella quale vuoi proseguire.**

Se stai usando un VIC 20 hai bisogno di memoria extra per questo gioco.

```

690 IF (U$="O" AND D=3 AND F(13)=0) THEN LET R=R-1:LET F(13)=1
700 IF (U$="E" AND D=4 AND F(13)=0) THEN LET R=R+1:LET F(13)=1
710 NEXT I

```

```

720 LET M$="OK"
730 IF F(13)=0 THEN LET M$="NON PUOI ANDARE PER DI LA'"
740 IF D<1 THEN LET M$="ANDARE DOVE?"
750 RETURN

```

Hai già provato a giocare alcune volte e hai deciso che ti sarà impossibile vincere? Potresti concederti più tempo, ponendo un valore maggiore per L alla linea 1910.



```

760 LET M$="TI STAI ALLENANDO PER LE OLIMPIADI?"
770 IF F(10)=1 AND (R=9 OR R=3) THEN LET M$="E' TROPPO DISTANTE PER SALTARE"
780 RETURN
□790 IF OB=8 THEN LET M$=V$+" IL "+W$+M2$:RETURN
800 IF OB>G THEN LET M$="NON POSSO RAGGIUNGERE "+W$:RETURN
810 IF L(OB)<O THEN LET M$="QUI NON C'E'"
820 IF F(OB)<O THEN LET M$="CHE "+W$+"?"
830 IF L(OB)=25 THEN LET M$="CE L'HAI GIA'"
□840 IF OB>O AND L(OB)=R AND F(OB)=0 THEN LET L(OB)=25:LET M$=M3$+W$
850 RETURN
860 IF L(OB)<25 THEN RETURN
870 LET M$="NO DAVVERO!":IF OB>1 THEN RETURN
880 PRINT "DOVE DEVO METTERE LA CHIAVE"
890 INPUT W$:IF W$>"" THEN GOSUB 440
□900 IF (OB=8 OR W$="SUL "+O$(8)) THEN LET M$=M4$:LET F(15)=1:LET L(1)=R
910 RETURN
□920 IF F(11)=0 AND L(7)=25 THEN LET F(11)=1:LET F(3)=0:LET L(3)=0:LET M$=M5$
□930 IF F(11)=1 AND L(7)=25 AND R=20 THEN LET F(9)=1:LET M$=M$+M6$
940 IF R=21 AND OB=24 THEN LET F(4)=0:LET M$="LA POLVERE SI POSA"
950 IF OB=16 THEN GOSUB 970
960 RETURN
970 LET M$="NIENTE DI INTERESSANTE"
980 IF OB=16 AND R=12 THEN LET F(2)=0:LET M$="E' UN MICRO - VCR"
990 IF OB=24 AND R=21 THEN LET M$="C'E' QUALCOSA DENTRO"
1000 IF OB=19 AND R=24 THEN GOSUB 1400
□1010 IF OB=23 AND R=12 THEN LET M$=M7$
1020 IF OB=7 THEN LET M$="CONTIENE UNA GRANDE MOSCA"
1030 IF OB=9 AND R=20 AND F(9)=0 THEN LET M$="GRAFFIA E MORDE!"
1040 IF OB=4 THEN GOSUB 1080
1050 IF OB=1 AND L(1)=25 THEN LET M$="C'E' INCISO SOPRA IL NUMERO '111'"
1060 IF OB=6 AND L(6)=25 THEN LET M$="C'E' UN GRANDE BOTTONE ROSSO"
1070 RETURN
□1080 LET M$=M8$:IF OB=4 AND L(OB)=25 THEN LET M$=M9$
1090 RETURN
1100 IF OB>5 THEN LET M$="NON POSSO LEGARE "+W$:RETURN
1110 IF L(5)<25 THEN RETURN
1120 LET W$="":PRINT "LEGARE IL FILO A COSA"
1130 INPUT W$
1140 IF W$>"" THEN GOSUB 440
1150 LET M$="NON POSSO LEGARLO A "+W$
□1160 IF OB=13 AND R=9 THEN LET F(5)=1:LET L(5)=0:LET M$=M8$
1170 RETURN

```

**760-1500: Queste sono subroutine per i verbi usati nel programma. Cerca le linee con RETURN per vedere dove finisce una e inizia l'altra.**

Puoi tralasciare le parole LET se vuoi, a meno che tu non stia usando uno Spectrum.

```


1180 IF OB=5 AND F(5)=0 THEN LET M$="NON E' LEGATO A NULLA!"
1190 IF OB=5 AND R=9 AND F(5)=1 THEN LET R=8:LET F(10)=0:LET M$="OK":RETURN
1200 IF OB=5 AND R=8 AND F(5)=1 THEN LET R=9:LET F(10)=1:LET M$="OK"
1210 IF OB=13 AND R=9 THEN LET M$="E' TROPPO LISCIO PER SCALARLO"
1220 IF OB=W AND R=19 THEN LET OB=30:GOSUB 550

```





Questo programma non è spiegato così estesamente come gli altri, in modo da non darti troppi indizi sul gioco. Puoi cercare di scoprire cosa facciano le varie linee di programma dopo aver giocato alcune partite.



```

1230 IF OB=W AND R=20 THEN LET OB=29:GOSUB 550
1240 RETURN
1250 IF OB<>6 OR L(6)<>25 THEN RETURN
1260 LET W$="":PRINT "PUNTARLO A COSA"
1270 INPUT W$
1280 IF W$>" " THEN GOSUB 440
1290 IF OB=22 THEN LET F(6)=1
1300 LET M$="MOLTO BENE"
1310 RETURN
1320 IF OB=3 AND L(OB)=25 THEN LET L(OB)=0:LET M$="MUNCH CHOMP"
1330 RETURN
▲1340 IF (OB=19 OR OB=17) AND R=24 AND L(1)<25 THEN LET M$=M9$:LET L=L-12:LET F(1
6)=1
1350 IF (OB=19 OR OB=17) AND R=24 AND L(1)=25 THEN LET F(12)=1:LET M$=MH$
1360 RETURN
▲1370 IF OB=3 AND R=7 AND L(OB)=25 THEN LET F(3)=1:LET L(OB)=0:LET B=B+1:LET M$=M
A$
1380 IF L(OB)=25 THEN LET L(OB)=R:LET M$="FATTQ"
1390 RETURN
1400 IF R<24 THEN LET M$="NON C'E' NULLA SU CUI SCRIVERE!":RETURN
1410 IF F(12)=0 THEN LET M$="QUESTO TERMINAL NON E' ATTIVATO":RETURN
1420 LET M$="IL TERMINAL FA ECO: "+W$
1430 IF OB=18 THEN LET M$="CODICE ACCETTATO":LET F(17)=1
1440 IF OB=20 THEN LET M$="IDENTIFICAZIONE TERMINAL"
1450 RETURN
1460 IF R=14 AND OB=11 AND L(8)=22 THEN LET L(8)=14:LET M$=MB$+MC$:RETURN
1470 IF R=14 AND OB=11 AND L(8)=14 THEN LET L(8)=22:LET M$=MB$+MD$
1480 IF F(15)=1 THEN LET L(1)=L(8)
▲1490 IF OB=21 AND F(6)=1 AND L(6)=25 AND R=18 AND F(17)=1 THEN LET B=B+10:LET M$
=ME$:LET F(14)=1
1500 RETURN

```

```

1510 LET S=0:FOR I=1 TO G
1520 IF L(I)=25 THEN LET S=S+1
1530 NEXT I
1540 LET S=S+B:PRINT "PUNTEGGIO = ";S
1550 IF F(14)=0 THEN PRINT "PREMI RETURN PER CONTINUARE":INPUT Q$
1560 RETURN

```

**1510-1560: Subroutine per il punteggio.**

```

1570 LET V=24:LET W=31:LET G=8
1580 DIM R$(24):DIM O$(W):DIM H$(V)
1590 DIM L(G):DIM F(17)

```

**1570-1590: Visualizza le variabili e dimensiona gli array.**

**1600-1870: Linee contenenti i dati.**

```

1600 DATA "NELLA TANA DEL TOPO - FA MOLTO BUIO QUI"
1610 DATA "PRESSO UNA TANA DI TOPO IN UN ANGOLO","SULL'ANGOLO DI UN ALTO TAVOLO"
1620 DATA "NEL RETRO DI UN'ENTRATA","IN UNO SGABUZZINO","NELLA CUCINA"
1630 DATA "PIU' GIU' IN UNA BUIA GALLERIA AMMUFFITA","VICINO A UN BINARIO MORTO"
1640 DATA "ALLA BASE DI UN ALTO ALBERO DI PLASTICA SULL'ANGOLO DI UN ALTO TAVOLO"
1650 DATA "FUORI DELLA PORTA SPALANCATA DI UNA CASA DALLE STRANE DIMENSIONI"
1660 DATA "IN UNA STANZA GIALLA","VICINO A UN TELEVISORE E UN REGISTRATORE"
1670 DATA "ALLA FINE DI UNA GALLERIA BUIA"
1680 DATA "VICINO A UN GROSSO INTERRUTTORE CONNESSO AI BINARI"
1690 DATA "LUNGO IL BINARIO TORTUOSO"
▲1700 DATA "ALLA FINE DELLA LINEA: IL BINARIO SCOMPARE IN UN BUCO NEL MURO"
1710 DATA "AI PIEDI DI UN MURTO FATTO DI GIGANTESCHI SCHERMI TELEVISIVI"
1720 DATA "IN PIEDI SULL'INGRANDITORE"
1730 DATA "SULLO SCAFFALE DI APPARATI DI DISTURBO: C'E' UNO SGABELLO VICINO"
1740 DATA "SU UN CURTO SGABELLO"

```

```

1750 DATA "SUL FONDO DI UNA SCATOLA ROVESCIATA PIENA DI COMPONENTI ELETTRONICI"
1760 DATA "PRESSO A UN BUCO NEL MURO DA CUI EMERGE UN BINARIO FERROVIARIO"
1770 DATA "ALLA BASE DI UNA SEDIA GIREVOLE"
1780 DATA "IN PIEDI SU UN TERMINAL DI COMPUTER CON UN BLOCCO DI SICUREZZA"
1790 DATA 16,12,3,21,5,17,19,14
1800 DATA "INVENTARIO", "ANDARE", "N", "S", "O", "E", "A", "B"
1805 DATA "SALTARE", "PRENDERE", "PRENDERE", "METTERE", "APRIRE"
1810 DATA "ESAMINARE", "LEGGERE", "LEGARE", "ARRAMPICARE", "PUNTARE"
1815 DATA "MANGIARE", "APRIRE", "LASCIARE", "SCRIVERE"
1820 DATA "PREMERE", "PUNTEGGIO", "SE", "SO", "SE", "SOE"
1825 DATA "OE", "SO", "NS", "NS", "NE", "NO"
1830 DATA "EO", "NO", "NS", "NE", "EO", "O", "S", "S"
1835 DATA "NEB", "EOAB", "EO", "EO", "NEO"
1840 DATA "NO", "CHIAVE", "CASSETTA", "FORMAGGIO", "CARTA"
1845 DATA "FILO", "TELECOMANDO"
▲1850 DATA "BOTTIGLIA", "TRENO", "GATTO", "PORTA", "INTERRUTTORE"
1855 DATA "GALLERIA", "ALBERO", "BUCO", "TOPO"
1860 DATA "VIDEO", "COMPUTER", "DROGHIERE", "TERMINAL"
1865 DATA "111", "BUTTONE", "INGRANDITORE"
1870 DATA "TELEVISORE", "SCATOLA", "NORD", "SUD", "OVEST"
1875 DATA "EST", "ALTO", "BASSO", "SGABELLO"

```

#### 1880-1900: Lettura dati

```

1880 FOR I=1 TO 24:READ D$:NEXT I:FOR I=1 TO 6:READ L(I):NEXT I
1890 FOR I=1 TO V:READ H$(I):NEXT I:FOR I=1 TO 24:READ R$(I):NEXT I
1900 FOR I=1 TO W:READ O$(I):NEXT I:LET F(10)=1:LET F(4)=1:LET F(2)=1

```

```

□ 1910 LET R=11:LET B=8:LET L=100:LET M$="TI SVEGLI...":LET MH$="TERMINAL ATTIVO"
□ 1920 LET M1$="NON PUOI RAGGIUNGERE LA CHIAVE":LET M2$="EH? MOLTO SPIRITOSO!"
□ 1930 LET M3$="ORA HAI ":LET M4$="MOLTO BENE!":LET M5$="NULLA DI INTERESSANTE"
□ 1940 LET M5$="UNA MOSCA SVOLAZZA VIA RONZANDO":LET M6$="IL TRENO SBUFFA "
□ 1950 LET M7$="E' SOLO UNA SCATOLA CON UNA FOTO SOPRA":LET M9$="* TAMPER *!"
□ 1960 LET M8$="E' BEN LEGATO.":LET M6$="E IL GATTO SI METTE A CORRERGLI DIETRO!"
□ 1970 LET M4$="IL TOPO SE NE VA CON QUELLO":LET M5$="CODICE TERMINAL="+O$(18)
□ 1980 LET M6$="IN VISTA E SI FERMA QUI":LET M5$="SCOMPARE NELLA GALLERIA"
□ 1990 LET M6$="IL RAGGIO INGRANDITORE FUNZIONANDO TI RIPORTA ALLA MISURA NORMALE"
2000 RETURN

```

#### 1910-2000: Prepara i messaggi.

## Linee di conversione

- ▲ 30 Sostituisci CLS con PRINT CHR\$(147)
- 30 Sostituisci CLS con HOME

### ...PER UTENTI DELLO SPECTRUM □

```

120 PRINT R$(R,I);";";
190 FOR I=1 TO LI-1
200 IF Q$(I)="" AND V$="" THEN LET V$=Q$(TO I-1)
210 IF Q$(I+1)<>" " AND V$<>" " THEN LET W$=Q$(I+1 TO):LET I=LI-1
235 IF LEN(V$)<7 THEN LET V$=V$+" ":GOTO 235
360 GOSUB 480*(VB=1)+550*(VB>1 AND VB<9)+760*(VB=9)+790*(VB=10 OR VB=11)+860*
(VB=12)+920*(VB=13)+970*(VB=14)
380 GOSUB 1080*(VB=15)+1100*(VB=16)+1180*(VB=17)+1250*(VB=18)+1320*(VB=19)
+1340*(VB=20)+1370*(VB=21)+1400*(VB=22)+1460*(VB=23)+1510*(VB=24)+1560*
(VB=25)
440 IF LEN(W$)<15 THEN LET W$=W$+" ":GOTO 440
445 FOR I=1 TO W
660 LET U$=R$(R,I)
900 IF (OB=8 OR W$="SUL "+Q$(8)+" " THEN LET M$="MOLTO BENE!":LET F(15)=1:
LET L(1)=R
1580 DIM R$(24,4):DIM O$(W,14):DIM H$(V,7)
1910 LET R=11:LET B=8:LET L=100:LET M$="TI SVEGLI..."

```


Linee 620,790,840,920,930,1010,1080,1160,1340,1350,1370,1460,1470,1490:

Sostituisci le stringhe dalla M1\$ alla MH\$ con il testo nelle linee dalla 1910 alla 1990.

Per esempio, 620 IF R=2 AND L(1)=25 AND D=3 THEN LET M\$="NON PUOI RAGGIUNGERE LA CHIAVE":RETURN

Tralascia le linee dalla 1910 alla 1990.

**NOTA IMPORTANTE**  
per gli utenti di computer Commodore. Se stai usando un C64 o un VIC 20, devi tralasciare le parole LET alle linee 400, 1340, 1370 e 1490. Se hai un C64, devi tralasciare tutti gli spazi alle linee 1490, 1700 e 1850.



# Risposte ai quesiti

Ecco le soluzioni di alcuni dei quesiti proposti nel libro.

## Teschi della Piramide

Per aggiungere testi, sostituisci 5 alla linea 260 con un numero più alto.

I limiti del numero da raggiungere vengono fissati alla linea 20. In questo programma, il numero verrà scelto a caso tra 10 e 39. Se sostituisci 30 con 40, avrai un numero compreso tra 10 e 49. Se sostituisci anche 10 con 20, otterrai un numero compreso tra 20 e 49.



## L'Abbraccio del Mostro

Per aumentare il numero di round, sostituisci il numero 11 con uno maggiore alle linee 260 e 550 (che deve essere lo stesso in entrambe le linee). Se aumenti il numero di round, ti occorreranno più possibilità di ricorrere al tasto P: dovrai quindi cambiare il 4 alla linea 440 con un numero più alto.

## Lo Squalo (Jaws)

Per far muovere più spesso il cacciatore, diminuisci il valore di U alla linea 30, ponendolo per esempio a 0.8.

## Streghe Volanti

Se non porti a termine il tuo compito, il programma andrà alle linee 240 o 250 per un messaggio. Modificalo come vuoi.

© Copyright per l'edizione originale Usborne Publishing Ltd — 1984  
© Copyright per l'edizione Gruppo Editoriale Jackson — 1985

Impaginazione: Cristina De Venezia  
Fotocomposizione: Composit - Pisa  
Stampa: Grafika 78

Il nome Usborne e il marchio  sono marchi registrati dalla  
Usborne Publishing Ltd., 20 Garrick Street, London WC2E 9BJ, England.

Tutti i diritti sono riservati. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.









# Speedy Computer

Il calcolatore è anche (o soprattutto?) una macchina divertente. Si può giocare con lui, gli si possono fare delle domande, lo si può usare per scrivere una poesia o per suonare.

Questa colorata serie di libri vi propone alcune delle cose più eccitanti che si possono fare con un calcolatore e vi spiega come farle.

Scritti in linguaggio chiaro e comprensibile a chiunque, arricchiti da una moltitudine di illustrazioni, questi libri rappresentano una spiritosa introduzione al mondo dei computer per chi comincia da zero.

## **GIOCHI di spionaggio brivido e mistero**

Spionaggio, mistero, terrore sul vostro video: più di venti giochi che vi faranno star svegli la notte.

## altri volumi di questa collana

### **ESPANSIONI DEL PERSONAL COMPUTER**

Espansioni, accessori, interfacciamento... tutti i "cosa " e i "come " per far diventare il vostro piccolo computer una vera "centrale operativa".

### **IL COMPUTER CON FANTASIA**

Progettate anche voi un "fantasy" dopo aver appreso giocando tutti i segreti del "sotterraneo del fato".

### **IL MISTERO DELLA MONTAGNA D'ARGENTO**

L'affascinante mistero della montagna d'argento vi sarà svelato in un'avventura mozzafiato.

